
 

 

 

АА..  СС..  ГГрроошшеевв  
 

 

 

 

 

 

ППррооггррааммммииррооввааннииее  

ннаа  яяззыыккее  MMiiccrroossoofftt  

VViissuuaall  BBaassiicc  

SSccrriippttiinngg  EEddiittiioonn 
  

  

  

Методические указания 

 к выполнению лабораторных работ 

 

 

 
 

 

 

 

 

Архангельск 
2009 

Федеральное агентство по образованию 
Архангельский государственный технический университет 

Институт информационных технологий 



2 

 

Рассмотрено и рекомендовано к изданию методической 

комиссией Института информационных технологий 

Архангельского государственного технического университета 

 __ ____________ 2008 г. 

 

 

 

 

Рецензент 

___________________________________________________ 

 

___________________________________________________ 

 

 

УДК  004.65 

ББК   32.973.26 - 018.2 

Г 89 

Грошев А.С. Программирование на языке Visual Basic Scripting Edition: 

Метод. указания к выполнению лабораторных работ. – Архангельск, 

2009. – 82 с. 

Изд-во Арханг. гос. техн. ун-та, 

Предназначена для студентов вузов, изучающих основы про-

граммирования на алгоритмических языках. 

Содержит лабораторные работы по основным разделам програм-

мирования на алгоритмическом  языке Visual Basic Scripting Edition.  

Особенность данного учебного пособия – наряду с рассмотрени-

ем основ построения программ, рассмотрены вопросы использования 

языка программирования для работы с информацией текстовой, число-

вой, типа даты и время и с объектами  WScript и OLE Automation. 

 

 

 

Ил. 36. Табл. 11. Библиогр. 1 назв. 

 

 

 

 

 

 

 

                        © А.С. Грошев, 2009 



3 

 

ОГЛАВЛЕНИЕ 

Лабораторная работа № 1. Создание простейшей программы. Работа с 

окнами сообщений и ввода данных........................................................... 4 

Лабораторная работа № 2. Типы данных. Константы. Переменные ... 13 

Лабораторная работа № 3. Массивы ....................................................... 19 

Лабораторная работа № 4. Операторы IF и CASE ................................. 23 

Лабораторная работа № 5. Операторы цикла Do и While ..................... 29 

Лабораторная работа № 6. Операторы цикла For и For Each ............... 34 

Лабораторная работа № 7. Процедуры и функции пользователя ........ 37 

Лабораторная работа № 8. Работа с числовой информацией ............... 41 

Лабораторная работа № 9. Работа со строковой информацией ........... 44 

Лабораторная работа № 10. Работа с информацией типа дата и время46 

Лабораторная работа № 11. Работа с логическими выражениями ...... 51 

Лабораторная работа № 12. Работа с объектами WScript ..................... 54 

Лабораторная работа № 13. Работа с информацией файловой системы

 ..................................................................................................................... 59 

Лабораторная работа № 14. Работа с информацией об ошибках ......... 68 

Лабораторная работа № 15. Работа с информацией локальной сети .. 70 

Лабораторная работа № 16. Работа с объектами Windows OLE 

Automation (Microsoft ActiveX) ................................................................ 71 

Лабораторная работа № 17. Использование скриптов на HTML-

страницах ................................................................................................... 74 

Приложение 1. Математические функции языка................................... 76 

Приложение 2. Строковые функции языка ............................................ 78 

Приложение 3. Функции работы с датой и временем ........................... 79 

Приложение 4. Константы даты и времени ............................................ 80 

Приложение 5. Логические функции и операторы языка ..................... 81 

Приложение 6. Методы и свойства объекта ADO.Recordset ................ 82 

Литература ................................................................................................. 85 

 



4 

 

Лабораторная работа № 1. Создание простейшей 
программы. Работа с окнами сообщений и ввода 
данных 

Программа на языке VBScript состоит из инструкций языка 

(statement) в виде текстовых строк.  

Несколько инструкций языка можно объединить в одну строку в 

текстовом файле программы с использованием разделителя строк – 

символа двоеточия ( : ) и наоборот, одну строку программы можно на-

писать на нескольких строках в тексте с использованием символа под-

черкивания ( _ ).  

В русском языке инструкции языка программирования обычно 

называют операторами языка, хотя это не совсем точно: операторами в 

английских первоисточниках называют символы для обозначения ма-

тематических, логических и строковых операций (=, +, -, /, and, or, eqv, 

& и пр.). Далее будет использоваться традиционная русская термино-

логия с использованием слова операторы для обозначения инструкций 

языка. 

Текст программы можно написать в любом простейшем редакто-

ре, сохраняющем файлы в кодировке ASCII или Юникод, например, в 

стандартных программах Windows Блокнот или WordPad. 

Существуют также специализированные редакторы, предназна-

ченные для написания в них программ. 

Их преимущества: 

1) выделение различных синтаксических конструкций языка 

программирования разным цветом; 

2) наличие справочной системы по операторам, функциям и 

объектам языка; 

3) возможность отладки программы путем задания точек ос-

танова, пошагового прохождения с просмотром значений 

переменных; 

4) возможность запуска интерпретируемых программ непо-

средственно из редактора. 

Всеми этими достоинствами для языка VBScript обладает лицен-

зионная программа VbsEdit. 

Редактор EmEditor Professional может быть настроен на выделе-

ние синтаксиса различных языков: Bat, C#, C++, CSS, HTML, Java, 

JavaScript, Pascal, Perl, PerlScript, PHP, Python, RHTML, Ruby, SQL, 

TeX, VBScript, Windows Script, x86 Assembler, XML, имеет возмож-

ность запуска скриптовых программ и открытия зарегистрированных 

приложений для различных типов файлов. Несколько меньшие воз-

можности имеет редакторы Aditor Pro, TextPad и др.  

Далее воспользуемся редактором Блокнот, как наиболее доступ-

ный. 

Для создания простейшей программы делаем следующее: 

1) запускаем Блокнот; 

2) пишем в Блокноте строку:  MsgBox "Привет!" (это имя функции с 

аргументом – текстовой константой между апострофами; имя 

MsgBox – сокращение от английского выражения Message Box, ко-

торое дословно можно перевести, как «коробка сообщений», в сис-

теме Windows – окно 

сообщений;  

3) сохраняем текстовый 

файл с именем 

Prg1.vbs; 

4) в свойствах файла в 

пункте Приложение 

проверяем, что для ра-

боты с ним задана про-

грамма Microsoft Win-

dows Based Script Host 

(см. рисунок 1.1). Если 

этого нет, нажимаем 

кнопку Изменить и 

выбираем в папке 

\Windows\system32\  

файл wscript.exe; 

5) двойным щелчком мы-

ши открываем файл.  

Результат работы 

этой программы – диало-

говое окно в системе Win-

dows (Windows-форма), 

показанное на рисунке 1.2. 

 

 

Рисунок 1.1. Окно свойств файла Prg1.vbs 

Рисунок 1.2. Пример работы простейшей 

программы на языке Vbscript, исполняемой 

системой Windows Based Script Host  

(wscript.exe) 

Microsoft Windows 

Based Script Host 

При выполнении этой программы используется стандартная 

функция языка VBScript для вывода сообщений в окно Windows со сле-

дующим полным синтаксисом (здесь и далее в описании синтаксиса в 

квадратных скобках  [  ] приводятся необязательные элементы, эле-

менты в скобках < > должны быть заменены конкретными значения-

ми):  

[<p> = ] MsgBox( <Сообщение>[,<Кнопки и значок>] _ 

        [,<Заголовок окна>][,<Справка, раздел>] )  

где назначение аргументов функции следующее: 

p – переменная, которой присваивается код нажатой кнопки; 

Сообщение – текст в диалоговом окне; 

Кнопки и значок – стандартные переменные (приведены далее в 

таблице 1.1), определяющие кнопки, значок и номер кнопки 

по умолчанию в окне (например, vbYesNoCancel + vbInforma-

tion + vbDefaultButton3 или 3+64+512);  

Заголовок окна – надпись на заголовке окна  (например, «Мое 

первое окно»); 

Справка, раздел – имя файла справки и идентификатор  раздела, 

связанного с данным окном. 

Аргументы функции следует писать в круглых скобках, если сле-

ва стоит переменная, которой присваивается значение, возвращаемое 

функцией, иначе аргументы следует писать за именем функции через 

пробел без скобок. 

Дополним текст в файле Prg1.vbs указанными дополнительными 

параметрами (текст пишем в одну строку или используем знак под-

черкивания  _  в конце первой строки для продолжения текста функции 

на следующей строке): 

MsgBox "Привет!", vbYesNoCancel + vbInformation _ 

 + vbDefaultButton3, "Мое первое окно","tst.hlp",1 

Окно запущенной программы будет иметь вид, показанный на 

рисунке 1.3. 

 Сообщение  Кнопки, значок и номер кнопки по умолчанию 

 Заголовок окна  Справка, ее раздел 

Функция MsgBox может возвращать значение нажатой в окне 

кнопки (например 6, если нажата кнопка Yes (Да), либо другие зна-

чения для кнопок vbNo, vbCancel и пр., см. далее таблицу 1.1).  

Таблица 1.1. Константы диалоговых окон 

Константа 
Зна-

че-

ние 
Описание 

vbOKOnly 0 Показана только кнопка ОК 
vbOKCancel 1 Показаны кнопки ОК и Отмена 

(Cancel) 
vbAbortRetryIgnore 2 Показаны кнопки Стоп (Abort), По-

втор (Retry) и Пропустить (Ignore) 
vbYesNoCancel 3 Показаны кнопки  Да (Yes), Нет (No) 

и Отмена (Cancel) 
vbYesNo 4 Показаны кнопки Да (Yes) и Нет (No) 
vbRetryCancel 5 Показаны кнопки Повтор (Retry) и 

Отмена (Cancel) 
vbCritical 16 Показан значок Stop Mark (знак стоп) 
vbQuestion 32 Показан значок Question Mark (знак 

вопроса) 
vbExclamation 48 Выводится значок Exclamation Mark 

(восклицательный знак) 
vbInformation 64 Показан значок Information Mark (ин-

формационнный знак) 
vbDefaultButton1 0 По умолчанию в окне выбрана первая 

кнопка 
vbDefaultButton2 256 По умолчанию в окне выбрана вторая 

кнопка 
vbDefaultButton3 512 По умолчанию в окне выбрана третья 

кнопка 
vbDefaultButton4 768 По умолчанию в окне выбрана четвѐр-

тая кнопка 
vbSystemModal 4096 Диалоговое окно выводится в модаль-

ном режиме и располагается сверху 

Рисунок 1.3.  

Пример 

использования  

функции MsgBox 

языка Vbscript с 

заданием набора 

кнопок и иконки 

окна 

всех других окон 
vbOK 1 Нажата кнопка ОК. 
vbCancel 2 Нажата кнопка Отмена (Cancel) 
vbAbort 3 Нажата кнопка Стоп (Abort) 
vbRetry 4 Нажата кнопка Повтор (Retry) 
vbIgnore 5 Нажата кнопка Пропустить (Ignore) 
vbYes 6 Нажата кнопка Да (Yes) 
vbNo 7 Нажата кнопка Нет (No) 

Для определения кода нажатой в окне MsgBox кнопки следует 

использовать следующий синтаксис функции: слева нужно написать 

переменную, которой будет присвоено возвращаемое функцией значе-

ние, далее следует написать символ присваивания (=) и справа от него 

функцию, у которой аргументы написаны в круглых скобках: 

btn = MsgBox("Привет!", vbYesNoCancel + vbInformation _ 

  + vbDefaultButton3, "Мое первое окно") 

Диалоговое окно будет иметь тот же вид, что и раньше (только 

без кнопки Справка , см. рисунок 1.3), но после нажатия кнопки в окне 

переменная btn будет иметь значение, соответствующее нажатой 

кнопке. 

Если Вы желаете написать в окне Сообщение и Заголовок ок-

на, пропустив второй аргумент (Кнопки и значок), после первого ар-

гумента следует поставить 2 (ДВЕ!) запятые: 

MsgBox "Сегодня я написал свою первую программу на VBS!" _    

   , , "Окно сообщений студента Вани Иванова" 

 

Еще одна функция языка, позволяющая открывать окно для ввода 

пользователем с клавиатуры строки текста (максимальная длина строки 

256 символов): 

[<p> = ] InputBox( <Сообщение>[,<Заголовок окна>] _ 

      [,<Стр.умолч.>][,X][,Y][, <Справка, раздел>]) 

где новые параметры функции (по сравнению с MsgBox): 

Стр.умолч. – строковое значение в поле ввода, которое будет пока-

зано по умолчанию при открытии окна; 

X,  Y – координаты левого верхнего угла окна в единицах twips (1440 

twips = 1 дюйм,  567 twips = 1 см) по отношению к левому 

верхнему углу экрана; если координаты не указаны, окно вы-

водится примерно в центре экрана. 

Здесь 2 запятые, т. к. пропущен аргумент <Кнопки и значок> 

Дополним программу вводом данных с клавиатуры в окне 

InputBox :  

S = InputBox(vbLF & "Напишите строку текста:", _ 

        "Окно ввода. Студент Ваня Иванов", , 4000, 2000) 

Kod = MsgBox (S, vbYesNoCancel+vbInformation, _ 

     "Окно сообщений") 

Функция InputBox в данном примере имеет строку <Сообще-

ние>, состоящую из двух частей: константы языка VBScript – vbLF – 

код перехода на следующую строку, и текстового значения "Напишите 

строку текста:"; эти две части объединены в одну строку с исполь-

зованием оператора конкатенации &. 

Сохраним текст, как новый файл Prg2.vbs, откроем его двойным 

щелчком мышкой и увидим окно функции InputBox (рисунок 1.4). В 

поле ввода этого окна напишем новый текст (длиной более 256 симво-

лов) и нажмем кнопку Да, после чего появится окно, показанное на ри-

сунке 1.5, в сообщении которого присутствует 256 символов значения 

переменной S (если в первом окне нажать кнопку Нет, сообщение в ок-

не MsgBox будет отсутствовать). 

 

Рисунок 1.4. 

Пример исполь-

зования  функ-

ции InputBox 

 

Рисунок 1.5. 

Окно функции 

MsgBox с по-

казом текста, 

написанного в 

поле окна 
InputBox  

Еще одно стандартное окно с более сложным синтаксисом – с ис-

пользованием метода Popup объекта WScript.Shell – отличается от 

MsgBox тем, что для него можно задать время существования, после 

чего оно закроется с возвращаемым кодом -1, если пользователь не на-

Здесь 2 запятые, т. к. пропущена 
<Стр.умолч.> 

жал какую-либо его кнопку (результат выполнения программы показан 

на рисунок 1.6): 

Set W = CreateObject("WScript.Shell") 

W.Popup "Окно закроется через 7 сек. или раньше, " _ 

     & vbLF & "если Вы нажмете кнопку в окне", 7 , _  

  "Окно Popup библиотеки WScript.Shell",vbExclamation 

  Полный синтаксис метода Popup объекта WScript.Shell: 

 [<p>=]<Экземпляр объекта WScript.Shell>.Popup _ 

      (<Сообщение> [,<Секунды ожидания>]  _ 

      [, <Заголовок окна>] [, <Кнопки и значок>)] 

 

 

 

 

Рисунок 1.6. Использова-

ние метода Popup объекта 
WScript.Shell 

В изложенном выше материале использовались такие осно-

вополагающие понятия языка программирования, как строковые значе-

ния (символы между апострофами), стандартные константы окон и 

строковая константа vbLF – код перехода на новую строку, перемен-

ные (btn, S, Kod), операции присваивания (=) и конкатенации (& – 

объединение двух выражений любого типа в одну строку),  функции с 

параметрами (MsgBox, InputBox, Now, CreateObject), объекты 

(WScript.Shell) и их методы.  

Последующие лабораторные работы посвящены освоению ис-

пользования всех этих компонентов языка VBScript. 

 

Задания 

Написать программу для варианта задания, соответствующего 

номеру Вашего компьютера. Выполнить программу, сохранить ее текст 

и скриншоты окон в отчете. Все окна должны иметь заголовки сле-

дующего вида: «Окно ввода (или сообщений) <Фамилия имя отчество 

студента>». 

1) Вывести в левый верхний угол экрана окно для ввода текстовой 

строки, показать эту строку в окне сообщений с одной кнопкой OK  

и значком Information Mark. Затем в  окне Popup показать код нажа-

той кнопки при выходе из предыдущего окна. 

2) Вывести на расстоянии 10 см по горизонтали и вертикали от левого 

верхнего угла экрана окно для ввода текстовой строки, показать эту 

строку в окне сообщений с кнопками Да и Нет и значком 

Exclamation Mark. Затем в  окне Popup показать код нажатой кнопки 

при выходе из предыдущего окна. Определить коды нажатия для 

всех кнопок. 

3) Вывести в центре экрана окно для ввода текстовой строки, показать 

эту строку в окне сообщений с кнопками Повтор и Отмена и знач-

ком Stop Mark. Затем в  окне Popup показать код нажатой кнопки 

при выходе из предыдущего окна. Определить коды нажатия для 

всех кнопок. 

4) Вывести примерно в правом нижнем углу экрана окно для ввода 

текстовой строки, показать эту строку в окне сообщений с кнопками 

Стоп, Повтор и Пропустить и значком Question Mark. Затем в  окне 

Popup показать код нажатой кнопки при выходе из предыдущего 

окна. Определить коды нажатия для всех кнопок. 

5) Вывести в центре экрана окно для ввода текстовой строки, показать 

эту строку в окне сообщений с кнопками OK и Отмена и значком 

Exclamation Mark. Затем в  окне Popup показать код нажатой кнопки 

при выходе из предыдущего окна. Определить коды нажатия для 

всех кнопок. 

6) Вывести примерно в правом нижнем углу экрана окно для ввода 

текстовой строки, показать эту строку в окне сообщений с кнопками 

Да, Нет и Отмена и значком Question Mark. Затем в  окне Popup по-

казать код нажатой кнопки при выходе из предыдущего окна. Опре-

делить коды нажатия для всех кнопок. 

7) Вывести на расстоянии 20 см по горизонтали и 15 см по вертикали 

от левого верхнего угла экрана окно для ввода текстовой строки, по-

казать эту строку в окне сообщений с кнопками Стоп, Повтор и 

Пропустить и значком Information Mark. Затем в  окне Popup пока-

зать код нажатой кнопки при выходе из предыдущего окна. Опреде-

лить коды нажатия для всех кнопок. 

8) Вывести примерно в правом верхнем углу экрана окно для ввода 

текстовой строки, показать эту строку в окне сообщений с кнопками 

Да, Нет и Отмена и значком Exclamation Mark. Затем в  окне Popup 

показать код нажатой кнопки при выходе из предыдущего окна. Оп-

ределить коды нажатия для всех кнопок. 

9) Вывести слева примерно в центре по вертикали экрана окно для 

ввода текстовой строки, показать эту строку в окне сообщений с 

кнопками Стоп, Повтор и Пропустить и значком Question Mark. За-

тем в  окне Popup показать код нажатой кнопки при выходе из пре-

дыдущего окна. Определить коды нажатия для всех кнопок. 

10)  Вывести справа примерно в центре по вертикали экрана окно для 

ввода текстовой строки, показать эту строку в окне сообщений с 

кнопками Да и Нет и значком Information Mark. Затем в  окне Popup 

показать код нажатой кнопки при выходе из предыдущего окна. Оп-

ределить коды нажатия для всех кнопок. 

Лабораторная работа № 2. Типы данных. Константы. 
Переменные 

Каждый язык программирования предназначен для обработки 

информации (данных) различных типов. Используемые типы  данных и 

методы их обозначения и обработки могут несколько различаться в 

различных языках. 

Типы данных определяют:  

• формат представления данных в памяти компьютера;  

• область или диапазон возможных значений;  

• множество допустимых операций, применимых к данным.  

В языке Microsoft Visual Basic Scripting Edition определен един-

ственный тип данных – Variant. Это специальный тип, который мо-

жет содержать в себе различные виды информации. Все функции языка 

также возвращают данные типа Variant.  

Различные категории информации, которая может содержаться в 

типе Variant,  называются подтипами. 

В таблице 2.1 приведены подтипы данных, которые могут со-

держаться в типе Variant. 

Таблица 2.1. Подтипы языка VBScript 

Подтип Описание 
Byte Целые числа в диапазоне от 0 до 255. 

Boolean Логические значения True или False. 
Integer Целые числа в диапазоне от -32768 до 32767.  
Long Целые числа в диапазоне от -2 147 483 648 до 

 2 147 483 647. 
Single Числа одинарной точности с плавающей точкой в диапазоне от 

-3.402823E38 до -1.401298E-45 для отрицательных значений; от 
1.401298E-45 до 3.402823E38 для положительных значений. 

Double Числа двойной точности с плавающей точкой в диапазоне от     
-1.79769313486232E308 до -4.94065645841247E-324 для отрица-
тельных значений; 4.94065645841247E-324 до 
1.79769313486232E308 для положительных значений. 

Currency -922 337 203 685 477.5808 до 922 337 203 685 477.5807. 
Date / 
(Time)  

Числа, которые представляют даты и время в диапазоне между 
1-01-100 0:0:0 до 31-12-9999 23:59:59. 

Object Содержит объект 
String Строка переменной длины, которая максимально может содер-

жать 2 миллиона символов. 
Empty Неинициализированное значение (0 для числовых переменных, 

строка нулевой длины ("") для строковых переменных). 
Null Содержит неверные для подтипа данные. 
Error Содержит номер ошибки. 

ms-help://MS.VSCC.v90/MS.MSDNQTR.v90.en/script56/html/vskeytrue.htm
ms-help://MS.VSCC.v90/MS.MSDNQTR.v90.en/script56/html/vskeyfalse.htm

Функция VarType возвращает информацию о том, как данные 

сохранены в типе Variant. Для преобразования одного подтипа в дру-

гой могут использоваться соответствующие функции (Cbyte, Cdate, 

CSng, CDbl и др.). 

Другие диалекты языка Visual Basic также имеют тип Variant, 

но наряду с ним могут определять переменные различных типов, таких 

же, как подтипы языка VBScript. 

В тексте программы могут использоваться числа, строки текста, 

даты и время, которые являются константами. 

Константа – некоторое неизменяемое значение в тексте про-

граммы. Константа может иметь имя (идентификатор). 

Для тех констант, которые используются часто, можно задать 

имена. Задание имен константам делает программы легко читаемыми. 

Для этого в любом месте текста программы можно использовать сле-

дующее описание: 

Const N = 1.15e-15 

Const FIO = "Иванов Иван Иванович" 

Const Data_r = #05-13-1988 06:30:00# 

Const Time_r = #06:30:00# 

Const Gorod = "Архангельск" 

Как видно из примера, для числовых констант разделителем це-

лой и дробной части является точка, можно использовать экспоненци-

альный вид чисел (1.15e-15 = 1.15×10
-15

). Значения строковых констант 

следует писать между двумя кавычками ("), даты и времени – между 

двумя знаками решетки (#). 

В языке VBScript существует достаточно большое количество 

предопределенных констант, которые сгруппированы по категориям: 

1. Date and Time Constants – определяют константы для дат и 

времени для функций работы с ними (см. Приложение 4); 

2. Date Format Constants – определяют форматы дат и времени 

(см. Приложение 4); 

3. MsgBox Constants – используются в функции MsgBox и других 

диалоговых окнах (см. таблицу 1.1);  

4. String Constants – определяют скрытые символы, используемые 

для манипуляции со строками: 

 vbCr  – возврат каретки (Chr(13), переход в начало следующей 

строки); 

 vbLf – новая строка (Chr(10)); 

 vbCrLf – новая строка (Chr(13) + Chr(10)); 

 vbNewLine – новая строка (Chr(10) или Chr(13) + Chr(10)); 

 vbNullChar – символ с нулевым значением (Chr(0)); 

ms-help://MS.VSCC.v90/MS.MSDNQTR.v90.en/script56/html/vsfctvartype.htm

 vbNullString – строка с нулевым значением (Chr(0)); 

 vbTab – горизонтальная табуляция (Chr(9)) ; 

 vbVerticalTab – вертикальная табуляция (Chr(11)) ; 

5. VarType Constants – определяют форматы для различных 

подтипов (vbEmpty, vbNull, vbInteger, vbLong, vbSingle, vbSingle, 

vbCurrency, vbDate, vbString, vbObject, vbError, vbBoolean, vbVariant, 

vbDataObject, vbDecimal, vbByte, vbArray); 

6. File Attribute Constants – атрибуты файлов (Normal, 

ReadOnly, Hidden, System, Volume, Directory, Archive, Alias, 

Compressed); 

7. DriveType Constants – типы дисковых устройств (Unknown, 

Removable, Fixed, Remote, CDROM, RAMDisk); 

8. File Input/Output Constants – типы ввода-вывода для файлов 

(ForReading, ForWriting, ForAppending); 

9. Color Constants – определяют 8 основных цветов (могут исполь-

зоваться в HTML-скриптах): 

 vbBlack, vbRed, vbGreen, vbYellow, vbBlue, vbMagenta, vbCyan, 

vbWhite; 

10. Comparison Constants – определяют тип операции сравнения: 

 vbBinaryCompare – двоичное сравнение; 

 vbTextCompare – текстовое сравнение. 

Переменная – имя (идентификатор) в программе, связанное с 

областью оперативной памяти компьютера, предназначенной для хра-

нения какой-либо информации, которая может изменяться во время ра-

боты программы. 

Все переменные в языке VBScript имеют один тип – Variant и во 

время использования могут хранить данные разных подтипов. 

Правила написания идентификаторов переменных, констант, на-

званий процедур, функций, объектов, их методов и свойств следую-

щие: 

1) идентификатор должен начинаться с латинской буквы; 

2) может состоять из латинских строчных и прописных букв, 

цифр и символа подчеркивания (последний лучше не исполь-

зовать, могут быть проблемы, т. к. этот же символ использует-

ся, как признак переноса строки); 

3) длина его – не более 255 символов; 

4) буквы в верхнем и нижнем регистре не различаются; 

5) он должен быть уникален в области определения. 

ms-help://MS.VSCC.v90/MS.MSDNQTR.v90.en/script56/html/vsconvartype.htm

Для объявления переменных могут служить выражения: 

 Dim X, Y, Z 

 Public A, B, C 

 Private X1, X2, X3 

Однако, переменные в языке Basic можно и не объявлять с по-

мощью этих описаний, достаточно написать в программе новый иден-

тификатор и присвоить ему значение, после чего транслятор будет 

знать, что это переменная, например: 

Z = 1.2345 

S = ”Строка текста” 

DT = #12-31-08# 

Разница между описаниями Dim, Public, Private сущест-

венна при использовании внутри программы подпрограмм и функций 

пользователя. В этом случае различается область действия переменных 

и массивов. 

Для описания Dim область действия различается в зависимости от 

места его расположения: 

1) на уровне программы – переменные доступны в основной про-

грамме и во всех еѐ подпрограммах (глобальные переменные); 

2) описание в подпрограмме (процедуре или функции) – пере-

менные используются только в подпрограмме, где они описаны, ини-

циализируются и используются при исполнении этой подпрограммы, 

после завершения ее работы уничтожаются (локальные переменные).  

Описания Public и Private используются только на уровне про-

граммы, переменные в этом случае доступны в основной программе и 

во всех еѐ подпрограммах. 

При отсутствии явного описания все переменные, используемые 

в головной программе, являются глобальными, отсутствующие в го-

ловной программе, но используемые в подпрограмме – локальные. 

Если в начале программы написать строку:  

Option Explicit 

использование переменных без их явного описания в выражениях Dim, 

Public и Private будет запрещено. Попытка использовать необъяв-

ленную переменную вызовет сообщение об ошибке при выполнении 

программы. 

Явное описание переменных способствует уменьшению количе-

ства ошибок при программировании, в частности, предотвращает кон-

фликты между переменными, используемыми в основной программе и 

в процедурах при одинаковых именах; позволяет выявить неверно на-

писанные имена предопределенных констант языка (иначе неверные 

имена просто игнорируются). 

Задания 

В вариантах заданий в скобках < > задано значение переменных, 

которые нужно получить, в скобках ( ) – номера констант Вашего зада-

ния. Значения переменных следует формировать из констант, пробелов 

и знаков препинания с использованием оператора конкатенации &. 

1. Задайте в программе 3 константы и присвойте им значения 1) Ваше-

го имени, 2) отчества, 3) фамилии. Создайте 2 переменные и при-

свойте им значения: первой –  <(3), (1), (2)>, второй –  <(1) (2) (3)>. 

Покажите константы и переменные в модальном окне MsgBox (каж-

дую в отдельной строке).  

2. Задайте в программе 3 константы и присвойте им значения Ваших 1) 

дня, 2) название месяца, 3) года рождения. Создайте 2 переменные и 

присвойте им значения: первой –  <(1).(2).(3)>, второй –  <Я родился 

(1) (2) (3) года .>. Покажите константы и переменные в модальном 

окне MsgBox (каждую в отдельной строке). 

3. Задайте в программе 4 константы и присвойте им значения данных 

Вашего адреса проживания: 1) город; 2) улица 3) номер дома; 4) но-

мер квартиры. Создайте 2 переменные и присвойте им значения: 

первой –  <(1), (2), (3), (4)>, второй –  <Я живу в  городе (1)  на ули-

це  (2) в доме (3), квартира (4).>. Покажите константы и переменные 

в модальном окне MsgBox (каждую в отдельной строке). 

4. Задайте в программе 4 константы и присвойте им значения данных о 

Вашей учебе: 1) учебное заведение; 2) специальность 3) курс; 4) 

группа. Создайте 2 переменные и присвойте им значения: первой –  

<(1),(2), (3), (4)>, второй –  <Я учусь в  (1)  на специальности  (2)  

курс (3), группа (4).>. Покажите константы и переменные в модаль-

ном окне MsgBox (каждую в отдельной строке). 

5. Задайте в программе 3 константы и присвойте им значения данных о 

Вашей учебе в школе: 1) населенный пункт; 2) № школы; 3) люби-

мый предмет. Создайте 2 переменные и присвойте им значения: 

первой –  <(1), (2), (3)>, второй – <Мой любимый предмет был (3), 

когда я учился в (2)-й школе города (или название другого типа на-

селенного пункта)  (1).>. Покажите константы и переменные в мо-

дальном окне MsgBox (каждую в отдельной строке). 

6. Задайте в программе 4 константы и присвойте им значения паспорт-

ных данных (придумать близкие к возможным): 1) серия; 2) №; 3) 

кем выдан; 4) дата выдачи. Создайте 2 переменные и присвойте им 

значения: первой –  <(1) – (2) , (3) (4)>, второй – <Паспортные дан-

ные: серия (1), номер (2), выдан (4) (3).>. Покажите константы и пе-

ременные в модальном окне MsgBox (каждую в отдельной строке). 

7. Задайте в программе 3 константы и присвойте им значения Ваших 

антропометрических данных: 1) рост в см; 2) вес в кг; 3) окружность 

груди; 4) талии; 5) бедер, в см. Создайте 2 переменные и присвойте 

им значения: первой –  <(1), (2), (3) – (4) – (5)>, второй – <Мой рост 

(1) см, вес (2),  окружность груди, талии и бедер (3) – (4) – (5) см >. 

Покажите константы и переменные в модальном окне MsgBox (каж-

дую в отдельной строке). 

8. Задайте в программе 3 константы и присвойте им названия предме-

тов Вашего сегодняшнего расписания: 1) 1-я пара; 2) 2-я пара 3) 3-я 

пара. Создайте 2 переменные и присвойте им значения: первой –  

<(1) – (2) – (3)>, второй – <8.20-9.05  9.10-9.55 – (1); 10.10-10.55  

11.00-11.45 – (2); 12.00-12.45 12.50-13.35 – (3)>. Покажите констан-

ты и переменные в модальном окне MsgBox (каждую в отдельной 

строке). 

9. Задайте в программе 3 константы и присвойте им названия окон 

VBScript: 1) MsgBox; 2) InputBox 3) Popup. Создайте 2 переменные 

и присвойте им значения: первой –  <Окна VBScript: (1),  (2), (3)>, 

второй – <Их назначение и особенности: (1) – (здесь написать на-

значение), (2) – (здесь об этом окне), (3) – (здесь особенности этого 

окна).>. Покажите константы и переменные в модальном окне 

MsgBox (каждую в отдельной строке). 

10.  Задайте в программе 6 констант и присвойте им названия подтипов 

данных VBScript для работы с числами. Создайте 6 переменных и 

присвойте им значения: <(подтип) – диапазон данных (здесь указать 

диапазон)>. Покажите константы и переменные в модальном окне 

MsgBox (каждую в отдельной строке). 

Лабораторная работа № 3. Массивы 

Переменная, которая хранит единственное значение, называется 

скалярной. Переменная может использоваться и для хранения серии 

чисел, такая переменная является массивом.  

Массив – переменная, предназначенная для хранения пронуме-

рованной серии значений (элементов массива). 

Скалярная переменная может использоваться без явного описа-

ния с использованием ключевых слов Dim, Public или Private, если в 

начале программы не присутствует директива Option Explicit.  

Массив обязательно должен быть описан перед его использованием. 
 

Для описания массивов используются те же операторы, что и для 

переменных, только после имени переменной в круглых скобках ука-

зывается количество индексов и их максимальное значение у элементов 

массива. 

Полный синтаксис этих описаний следующий: 

Dim varname[([subscripts])][, varname[([subscripts])]]... 

Public varname[([subscripts])][,varname[([subscripts])]]... 

Private varname[([subscripts])][,varname[([subscripts])]]... 

где: 

Varname – имя переменной; 

Subscripts – имеет формат:  индекс1[,индекс2]... – мак-

симальные значения индексов, минимальное значение 

равно 0; массив может быть одномерный, двумерный 

и т. д. до 60. 

Пример: 

Dim X(99), Y(24,24), Z(99,99,99) 

где: 

X(99) – одномерный массив из 99 элементов;    

Y(24,24) – двухмерный массив размерностью 24×24  элемента; 

Z(99,99,99) – трехмерный массив размерностью 99×99×99. 

Массив в языке VBScript после его объявления имеет тип Va-

riant, поэтому его элементам можно присваивать значения различных 

типов. Максимальный размер массива ограничен размером свободной 

виртуальной памяти операционной системы. 

Массив может быть динамический (изменяемой размерности), 

при его описании в круглых скобках размерность не указывают: 

Dim varname() 

Для инициализации динамического массива следует использовать 

оператор: 

ReDim [Preserve]varname(subscripts)[,varname(subscripts)]... 

Параметр Preserve используется, если выполняется повторная 

инициализация для изменения размера массива и необходимо сохра-

нить значения переменных, которые им уже были присвоены. 

Пример: 

Dim X() 

ReDim X(10, 10, 10) 

. . . 

ReDim Preserve X(10, 10, 15) 

Разница между описаниями Dim, Public, Private сущест-

венна при использовании внутри программы подпрограмм и функций 

пользователя. В этом случае различается область действия переменных 

и массивов. 

Для описания Dim область действия различается в зависимости 

от места его расположения: 

1) на уровне программы – переменные доступны в основной про-

грамме и во всех еѐ подпрограммах и функциях; 

2) описание в подпрограмме или функции –  переменные использу-

ются только в подпрограмме, где они описаны, инициализируют-

ся и используются при исполнении этого модуля, после заверше-

ния его работы уничтожаются.  

Описания Public и Private используются только на уровне ос-

новной  программы, переменные в этом случае доступны в этой про-

грамме и во всех еѐ подпрограммах и функциях. 

Другой способ создания переменной типа Variant, содержащей 

одномерный массив, –  с помощью функции Array: 

A = Array(10,20,30,40) 

В этом случае значения элементов массива будут следующими: 

A(0)=10, A(1)=20, A(2)=30 и A(3)=40. 

Пример программы с использованием различных описаний пере-

менных и массивов: 

Option Explicit 

Dim  Y(99), Z(99,99,99) 

Y(0) = "Это 1-ый элемент массива Y - строка" 

Y(1) = 123.456        ' Тип второго элемента - число  

Y(99)= #12-30-2007# ' Тип 100-го элемента - дата 
Z(99,99,99)="Это элемент трехмерного массива Z "_ 

            & "с индексом 99,99,99" 

MsgBox Y(0) & vbLf & "2-ой элемент массива Y - число: " _ 

     & Y(1) & vbLf & "100-ый элемент массива Y - дата: " _ 

     & Y(99)& vbLf & Z(99,99,99),,"Окно 1" 

Dim X     'Глобальная переменная X 
X="Это переменная X основной программы" 

MsgBox X,,"Окно 2" 

Call my_proc 

 Sub my_proc 

  Dim X   'Локальная переменная X 
  X="Это переменная X подпрограммы" 

  MsgBox X,,"Окно 3" 

 End Sub 

MsgBox X,,"Окно 4 – снова глобальная X" 

       Окна, которые показывает эта программа, показаны на рисунке 3.1. 

 
 

  

Рисунок 3.1. Использование различных описаний переменных и массивов 

Задания 

1) Опишите в программе два одномерных массива размерностью 3 и 5 

элементов, присвойте значения элементам первого массива – нечет-

ные числа, начиная с 21, второго – буквы русского алфавита, начи-

ная с мягкого знака. Покажите все данные в окне сообщений.   

2) Опишите в программе динамический массив. Выполните вначале 

его инициализацию для размерности 3 элемента и присвойте значе-

ния элементам массива – любые числа. Покажите все данные в 1-м 

окне сообщений.  Затем выполните повторную инициализацию для 

размерности 7 с сохранением значений определенных ранее элемен-

тов. Присвойте элементам с 4 по 7-й любые даты. Покажите все 

данные во 2-м окне сообщений. 

3) Опишите в программе двумерный массива размерностью 2 × 3 эле-

ментов и присвойте значения каждому элементу массива – время в 

диапазоне от 7:00 до 19:00. Покажите данные в окне сообщений в 

виде матрицы, в которой номер строки – первый индекс, в строке 

изменяется второй индекс.  

4) Задайте с помощью функции  Array значения 5-ти элементам масси-

ва, представляющим собой геометрическую прогрессию. Покажите 

все данные в окне сообщений. 

5) Создайте с помощью функции  Array одномерный массив, состоя-

щий из 6-ти чисел,. Покажите данные в окне сообщений. С помо-

щью функции ReDim переопредилите его размерность до двумерно-

го размерностью 3 × 2. Задайте значения всем его элементам и по-

кажите их в окне сообщений в виде матрицы, в которой номер стро-

ки – первый индекс, в строке изменяется второй индекс. 

6) Опишите в программе трехмерный массива размерностью 2 × 3 × 4 

элементов и присвойте числовые значения элементам массива. По-

кажите данные в окне сообщений с указанием элемента массива и 

его значение (например, A(0,0,0)=1 и т. д.). 

7) Задайте элементам двумерного массива текстовые значения – каж-

дому одно слово какого-либо четверостишия. Покажите элементы 

массива в окне сообщений в виде стихотворения. 

8) Опишите в программе два одномерных массива X и Y размерностью 

5 элементов, присвойте числовые значения элементам массивов. 

Покажите данные в окне сообщений в виде таблицы, в первой стро-

ке которой показаны имена массивов, в последующих – значения их 

элементов. 

9) Опишите в программе одномерный массив из 7 элементов. При-

свойте значения элементам – целые числа. Покажите элементы мас-

сива в окне сообщений в следующем порядке: 1, 7, 2, 6, 3, 5, 4. 

10)  Опишите в программе два одномерных массива размерностью 5 

элементов, присвойте числовые значения элементам массивов. По-

кажите данные в окне сообщений: в первой строке элементы перво-

го массива от первого до 5-го, во второй строке – элементы второго 

массива от 5-го до первого. 

Лабораторная работа № 4. Операторы IF и CASE 

Оператор условного перехода IF позволяет выполнить те или 

иные строки программы в зависимости от логических условий.  

В языке VBScript он может использоваться в двух различных ви-

дах (строчный и блочный синтаксисы). 

Строчный синтаксис: 

If <условие> Then <операторы1> [Else  <операторы2>] 

где: 

условие – логическое выражение, результатом вычисления ко-

торого может быть истина (True), ложь (False) или Null ко-

торое приравнивается к False; 

операторы1 – один оператор или более (разделенных двоето-

чиями для строкового синтаксиса); выполняются, если усло-

вие истинно (True); 

операторы2 – выполняются, если  условие не является истиной 

(False). 

В логических выражениях могут использоваться следующие ос-

новные операторы сравнения и логические операции (описание исполь-

зования последних см. в Приложении 5): 

= Равно  And Логическое «И» 

<> Не равно  Or Логическое «ИЛИ» 

< Меньше  Xor 

 
Логическое исключение 
(E1 Xor E2 возвращает True, если 
только E1 = True или только 
E2 = True, иначе – False) 

> Больше  

<= Меньше или равно  

>= Больше или равно  Eqv Логическое «эквивалентно» 

Is Сравнение объектов  
Imp 

Логическая импликация 
(E1 Imp E2 возвращает False, если 
E1 = True и E2 = False, иначе –True) Not Логическое отрицание  

Для простых условных операторов следует использовать строч-

ный синтаксис.  

Пример строчного синтаксиса:  

If A <= 9 Then A = A + 1 : B = B + A Else B = B + A 

Блочный синтаксис является более структурированным, имеет 

большие возможности, легче читается и отлаживается. В одном опера-

торе может быть выполнена проверка нескольких условий с заданием 

различных исполняемых фрагментов программы.  

Блочный синтаксис оператора условного перехода: 

If <условие> Then  

    [операторы]  

  [ElseIf <условие-n> Then  

    [операторы-n]] ... 

  [Else   

    [else-операторы]]  

End If 

где: 

условие – логическое выражение, результатом вычисления ко-

торого может быть истина (True), ложь (False) или Null ко-

торое приравнивается к False; 

операторы – один оператор или более (разделенных двоеточия-

ми для строкового синтаксиса), которые выполняются, если 

условие истинно (True); 

условие-n – то же, что и условие; 

операторы-n – выполняются, если  условие-n является исти-

ной (True); 

else-операторы – один оператор или более, выполняющиеся, 

если предшествующие условия не были истинны.  

Когда выполняется блочный If, проверяется условие, и, если 

оно истинно (True), выполняются операторы, следующие за Then. 

Если условие не является истинным (False), каждое условие-n, 

идущее за ElseIf (если они есть) проверяется. Когда истинное значе-

ние найдено, выполняются операторы-n, следующие за Then после 

истинного условия, после чего программа выходит за End If (т. е. по-

следующие ElseIf, если они есть, не проверяются). Если истинных 

условий для ElseIf не найдено, выполняются  else-операторы, 

следующие за Else. 

Пример блочного синтаксиса: 

FIO="Лютикова Лилия Максимовна" 

a=InputBox("Задайте значение переменной a","Пример IF. 

"&FIO) 

a=Eval(a) 'преобразование строки в число  

If a > 10 Then 

    b = "a > 10"   

  ElseIf a > 0 Then 

    b = "a > 0"  'будет выполнено только это при а=1! 

  ElseIf a = 1 Then 

    b = "a = 1" 

  Else 

    b = "Нет данных для заданного значения a" 

End If 

file:///C:\ASG\�������\Informatika\���_������_�����������\����������������\VBScript\Microsoft%20VBScripr%20Reference\glossary.htm%23defNull

MsgBox "Результат выполнения IF для a = "& a & _ 

       ": " & b,,FIO 

Пример использования оператора условного перехода:  

расчет суммы страховых взносов в Пенсионный фонд РФ, где 

суммы перечислений (Sp и Np) зависят от налоговой базы (S) и года 

рождения (G): 

S 
G < 1967 G >= 1967 

Sp  Np Sp  Np 

S <= 280000 10,3•S/100 0 4,3•S/100 6,0•S/100 

280000<S<=600000 
28840+5,5(S–

280000)/100 
0 

12040+3,1(S

–

280000)/100 

16800+2,4(S 

–280000)/100 

S>600000 46440 0 21960 24480 

Dim Sm, G, Sp, Np 

S=InputBox("Укажите налоговую базу", _ 

           "Расчет взносов в ПФ") 

s=Eval(s) 'преобразуем строковое значение в число 
G=InputBox("Укажите год рождения", _ 

           "Расчет взносов в ПФ") 

Np = 0 : Sp = 0 

If G < 1967 Then   ‘расчет для G < 1967 
 If S < 280000 Then  

          Sp = 10.3*S/100  

   ElseIf S > 280000 and S <= 600000 Then 

          Sp = 28840+5.5*(Sm-280000)/100 

   Else  

          Sp = 46440 

     End If 

 Else      ‘далее расчет для G >= 1967 
 If S < 280000 Then 

        Sp = 4.3*S/100 : Np = 6.0*S/100 

   ElseIf S > 280000 and S <= 600000 Then  

   Sp = 12040+3.1*(S-280000)/100 

   Np = 16800+2.4*(S-280000)/100 

   Else  

         Sp = 21960 : Np = 24480 

    End If 

End If   

MsgBox "Np =" & Sp & "  Sp = " & Np 

Оператор выбора Case позволяет выполнить те или иные опера-

торы в зависимости от множества значений заданного выражения или 

переменной. 

Синтаксис оператора выбора: 

Select Case <тест-выражение> 

   [Case <список_выр-n> 

      [<операторы-n>]] . . . 

   [Case Else 

      [<else-операторы-n>]] 

End Select 

где: 

тест-выражение  – любое числовое или строковое выражение; 

список_выр-n  – список из одного или более выражений для 

соответствующего Case; 

операторы-n  –  один оператор или несколько, выполняющихся, 

если тест-выражение имеет то же значение, что и значение 

одного из выражений списка-n; 

else-операторы-n  –  один оператор или несколько, выпол-

няющихся, если тест-выражение не совпадает ни с одним 

из значений Case-структур. 

Существует ряд особенностей в выполнении структуры Select 

Case: для целых чисел условие отбора сработает и для соответствую-

щего строкового подтипа, но для действительных чисел такое соответ-

ствие не наблюдается. как показано в следующем примере: 

A = 1 

Select Case A 

 Case 1.1, 1.2, 1.3 MsgBox "A = 1.1, 1.2 или 1.3" 

 Case "0.5","1","1.5" MsgBox "A = ""0.5"",""1"",""1.5""" 

     'будет выполнено только для Case "0.5","1","1.5"! 

 Case 0.5,1,1.5 MsgBox "A = 0.5,1.55,1.56" 

 Case Else MsgBox "Нет данных" 

End Select 

A = 1.55 

Select Case a 

   Case 1.1,1.2,1.3  MsgBox "A = 1.1,1.2,1.3" 

   Case "0.5","1.55","1.56"  MsgBox "A = ""0.5..."""  

   Case 0.5,1.55,1.56 MsgBox "A = 0.5,1.55,1.56" 

       'будет выполнено только для Case 0.5,1.55,1.56! 

   Case Else MsgBox "Нет данных" 

End Select 

Если же определить переменную a = "1.55" (строковое значе-

ние), в приведенном примере возникнет ошибка при выполнении с со-

общением о несоответствии типов. 

Задания 

1) Для Вашего варианта таблицы 4.1 задайте в окне ввода значение пе-

ременной X с учетом заданного подтипа данных.  

При вводе маленьких или больших чисел с использованием буквы e 

(например, -1e15) используйте преобразование подтипа строка в число 

с использованием функции Eval.  

При вводе даты и времени используйте функцию преобразования 

подтипа Cdate. 

При работе с датами учитывать, что их основные форматы 

#мм/дд/гггг# или #мм-дд-гг#, однако при написании года двумя циф-

рами и наличии на первом месте числа больше 12, формат преобразует-

ся в #гггг-мм-дд# (см. далее Лабораторную работу № 10). 

Вычислите переменную Y по одному из выражений в зависимости 

от значения X. Значения переменных X и Y покажите в окне сообще-

ний.  

Выполнить данное задание с использованием: 

a) строчного синтаксиса оператора условного перехода, 

b) блочного синтаксиса оператора условного перехода. 

Таблица 4.1. Варианты заданий 

№ Условие Y  № Условие Y 

1.1 X ≤ -10
15

 Y= «маленькое 

число» 

 1.6 X – месяц от 1 

по 3 

Y= «1-й квар-

тал» 

X>-10
15

 и X<0 Y= «отрица-

тельное число » 

 X – месяц от 4 

по 6 

Y= «2-й квар-

тал» 

X≥0 и X<10
15

 Y= «положи-

тельное число » 

 X – месяц от 7 

по 9 

Y= «3-й квар-

тал» 

X ≥ 10
15

 Y= «большое 

число» 

 X – месяц от 

10 по 112 

Y= «4-й квар-

тал» 

1.2 X – символ до 

«Г» 

Y= 1  1.7 X < -10
308

  Y= -∞ 

X – символ от 

«Г» до «Ж»  

Y= 2  X ≥ -10
308

 и 

X≤ 10
30

   

Y= «диапазон 

действитель-

ных чисел»  

X – символ по-

сле «Ж» 

Y= 3  X > 10
308

 Y= +∞ 

1.3 X – дата 

меньше 

01.01.1900 

Y= 19  1.8 X от 0 по 255 Y = «подтип  

Byte» 

X – дата от 

01.01.1900 до 

31.12.1999 

Y= 20  X от -32768 по 

32767 

Y = «подтип  

Integer» 

X – дата начи-

ная с 

01.01.2000 

Y= 21  X – целые 

числа другие  

Y = «подтип  

Long» 

1.4 X – время от  0 

час. 00 мин. до 

6 час. 00 мин. 

Y= «ночь»  1.9 X – дата и 

время  =  

1.1.2010 0:0:0 

Y = «С Но-

вым годом!» 

X – время от 6 

час. 01 мин. до 

12 час.00 мин. 

Y=   «утро»  X – дата от 1 

января 0000 

года по 31 де-

кабря 2099 

Y = «21 век!» 

X – время от 

12 час. 01 мин. 

до 18 час. 00 

мин. 

Y= «день»  X – дата от 1  

января 1900 

года по 31 де-

кабря 0099 

года 

Y = «20 век!» 

X – время от 

18 час. 01 мин. 

до 23 час. 59 

мин. 

Y= «вечер»   X – дата от 1 

января 100 

года по 

31.12.9999 

Y = «верный 

диапазон 

дат!» 

1.5 X – месяц от 

12 по 2 

Y= «зима»  1.10 X<0 Y= «X отри-

цательное» 

X – месяц от 3 

по 5 

Y=   «весна»  X≥0 и X<10
-15

 Y= «X ма-

ленькое по-

ложительное 

число» 

X – месяц от 6 

по 8 

Y= «лето»  X≥ 10
-15 

и X<1 Y= «X мень-

ше 1» 

X – месяц от 9 

по 11 

Y= «осень»  X≥ 1  Y= «X не 

меньше 1» 

 

2) С использованием оператора выбора Case выполнить задания 1-го 

пункта: 

для компьютеров с № 1 по № 5 – вариант   1.5, с № 6 по № 10  – ва-

риант 1.6. 

Лабораторная работа № 5. Операторы цикла Do и 
While 

Оператор цикла позволяет выполнить группу операторов не-

сколько раз в соответствии с заданными условиями повтора.  

Существует несколько видов оператора цикла:  

1) Do... Loop 

2) While ... Wend 

3) For ... Next 

4) For Each ... Next.  

Данная лабораторная работа посвящена первым двум.  

Синтаксис оператора Do...Loop следующий  

(здесь и далее в фигурных скобках {} приведены два возможных вари-

анта, разделенных вертикальной чертой |, один из которых необходимо 

использовать):  

1) первый вариант – проверка условия в начале цикла 

Do [{While | Until} <условие>] 

   [<операторы>] 

   [Exit Do] 

   [<операторы>] 

Loop  

2) второй вариант – проверка условия в конце цикла 

Do 

   [<операторы>] 

   [Exit Do] 

   [<операторы>] 

Loop [{While | Until} <условие>]  

где:  

условие – логическое выражение, которое имеет значение исти-

на (True) или ложь (False); значение условия Null то же, что 

и False;  

    для While (англ. пока) выполнение цикла продолжается, по-

ка условие истинно,  

        для Until (англ. до) – выход из цикла, когда условие истинно; 

операторы – один или несколько операторов, выполнение кото-

рых повторяется, пока условие после while истинно (True) 

или условие после Until ложно (False); 

Exit Do – может использоваться, как альтернативный выход  из 

цикла (на следующую строку программы после Loop); лю-

бое количество Exit Do может быть помещено внутри 

file:///C:\ASG\�������\Informatika\���_������_�����������\����������������\VBScript\Microsoft%20VBScripr%20Reference\glossary.htm%23defNull

цикла. Обычно эта команда используется с вычисляемым ло-

гическим выражением оператора If...Then. 

Пример использования операторов циклов Do While...Loop и 

Do Until...Loop: 

eps = 1e-7 

a = 1 

s = a 

n = 2 

Do While Abs(a) > eps  ' цикл выполняется, пока |a| > eps  

       ' или  Do Until Abs(a) <= eps, что аналогично предыдущему 
  a = - a * (2*n - 3)/(2*n - 1) 

  s = s + a :  n = n + 1 

Loop 

MsgBox("Расч. Pi = " & 4*S & vbLf & "n = " & n) 

Результатом работы программы будет число  = 

3,14159285358975 при n=5000002, продолжительность расчета менее 

30 сек. ( = 3,1415926535897932384626433832795 на калькуляторе 

Windows). Если задать eps = 1e-8, то время расчета увеличится до 3 

– 4 мин., результат расчета  = 3,14159267359025 при n=50000002.  

Внимание! При неверном написании условий окончания цикла про-

грамма может зациклиться (будет работать бесконечно долго). Чтобы пре-

кратить выполнение зациклившейся программы, необходимо открыть сред-

ство Windows Диспетчер Задач (Task Manager) с использованием сочетания 

клавиш Ctrl+Alt+Delete или, щелкнув правой кнопкой мыши на пустом месте 

панели задач, и выбрав в контекстном меню это средство, далее в разделе 

процессов найти и выделить wscript.exe и нажать кнопку Завершить процесс 

(End Process). 

Пример использования операторов циклов Do...Loop While и 

Do...Loop Until (результат работы – рисунок 5.1): 

Randomize 

Otvet1 = vbYes 

Otvet2 = vbNo  

Do 

 Do  
   MySum = FormatNumber(1000000*Rnd,2) 

        'Случайное число от 0.00 до 1000000.00 
   SSum = "Случайная сумма = " & MySum 

   Otvet2 = MsgBox(SSum & " руб." & vbLf & vbLf _ 

            & " Вам нравится такая сумма?", _ 

            vbYesNo,"Максимум - 1 000 000 руб.!") 

 Loop Until Otvet2 = vbYes  

              'выход из цикла, если нажата кнопка Да 

 proc = FormatNumber(MySum/1000000*100,1) 

 s = MySum & " руб. составляет " & proc & " % от 1 

млн."  

 If proc>95 Then  

  s = s & vbLf & " Редко бывает больше!" 

 ElseIf proc>85 Then  

 s = s & vbLf & " Совсем неплохо!" 

 ElseIf proc>75 Then  

 s = s & vbLf & " Бывает и больше...!" 

 Else  

  s = s & vbLf & " Маловато!" 

 End If 

 Otvet1 = MsgBox(S & vbLF & vbLF & _ 

   "Повторить генерацию случайных чисел?", _ 

    vbYesNo,"Оценка результата... ")  

Loop While Otvet1 = vbYes  

       ' цикл повторяется, если нажата кнопка Да 

 
 

Рисунок 5.1. Пример использования циклов  Do...Loop  

Синтаксис оператора цикла While...Wend следующий: 

While <условие> 

    [<операторы>] 

Wend 

Выполнение операторов цикла повторяется, пока <условие> ис-

тинно (True).  

Пример использования оператора While...Wend для расчета 

значения y = arcctg(x) с использованием итерационного ряда: 



eps = 1e-28 

k = 0 : s = 0  : a = 1 : z = 1 

'x = 0.9 'вариант для предварительного тестирования 
x = 0.9999999 

t1=Time 

Set W = CreateObject("WScript.Shell") 

While Abs(a) > 1E-14 

  a = z*x^(2*k+1)/(2*k+1) 

  s = s + a 

  k = k + 1 

  z = -z 

  If  Int(k/1e6)=k/1e6 Then 

      W.Popup "k = " & k & " Y = " & s, 1 

  End If  

Wend 

t2=Time  

MsgBox "Истинное значение  "& Atn(x) & vbLf & _ 

       "Расчетное значение "& s & vbLf &_ 

      "Погрешность по a = " & FormatNumber(Abs(a),15)_ 

      & vbLf & "Разница расч.- ист.= " & _ 

      FormatNumber(Abs(Atn(x)-s),15) & vbLf &_  

      "Продолжительность расчета " & _ 

      FormatDateTime(t2-t1) & vbLf &_ 

      " k = " & k , vbExclamation,_ 

      "Y = arcctg(x). Вариант 1" 

Результат выполнения этой программы показан на рисунке 5.2. 

 

 

Рисунок 5.2. Результаты работы программы с циклом  While... Wend 

Задания 

Вычислите число  по итерационной формуле с номером, соот-

ветствующим номеру Вашего ПК, с абсолютной погрешностью вычис-

ления от 10
-5

 до 10
-16

. Найдите величину погрешности, при которой в 

числе  постоянными остаются 7 знаков после запятой. Определите 

программно время расчета для каждого варианта, покажите в окне со-

общений таблицу, показывающую расчетное значение , n и продолжи-

тельность расчета для различной погрешности. 

Напишите пять вариантов программы для цикла DO с проверкой 

условия в начале и в конце и для цикла WHILE. Расчет для максималь-

ной точности выполнить для одного варианта, т. к. его продолжитель-

ность может составлять 15 – 30 мин.  

1) 1 
 

2)  

 

3) 2 

 

4)  

 

5) 3 

 

6)   

 

7) 4 

 

8) 5 

 

9) 6 

 

10) 7 

Лабораторная работа № 6. Операторы цикла For и 
For Each 

Синтаксис оператора цикла For...Next следующий: 

For <счетчик> = <нач.знач.> To <кон.знач.> [Step <шаг>] 

    [<операторы>] 

    [Exit For] 

    [<операторы>] 

Next 

где: 

счетчик – числовая переменная, используемая как счетчик цикла; 

может быть положительной или отрицательной величиной 

нач.знач. – начальное значение счетчика; 

кон.знач. – конечное значение счетчика; 

шаг – шаг изменения счетчика; на данную величину автоматически 

изменяется счетчик после каждого выполнения операторов 

цикла; если шаг не указан,  значит он равен 1; 

операторы – выполняются повторно столько раз, сколько определе-

но значениями, заданными для счетчика: один раз, много раз 

или ни одного; 

Exit For – может использоваться, как альтернативный выход из 

цикла;  обычно используется с проверкой условия выхода в опе-

раторе If...Then; выход выполняется на строку программы, 

следующую за Next. 

Пример использования цикла For...Next (см. рисунке 6.1):  

S="" 

For i = 1 to 5.5 step 1/5.5    

 s = s & i & vbLF 

Next 

MsgBox S,, " For ... step 1/5.5"  

Пример программы с альтернативным выходом (см. рисунок 6.2):  

S = "    X        Y" & vbLF  

S = S & "---------------" & vbLF 

For X = 1 to 5 step 0.11 

  Y = FormatNumber(Tan(X),3) 

    If Abs(Y) < 0.1 Then 

   s = s & "---------------" & vbLf _ 

       & "Выход из цикла" & vbLf & "при |Y| < 0.1" 

   Exit For 

    End If 

    s = s & FormatNumber(X,3) & "    " & Y & vblf 

Next 

MsgBox S,," For...Exit For...Next" 

Рисунок 6.1. 
For...Next 

Рисунок 6.2. For...Exit 
For ...Next 

Рисунок 6.3. Цикл  
For Each 

Синтаксис оператора цикла For Each...Next следующий: 

For Each <элемент> In <группа> 

    [операторы] 

    [Exit For] 

    [операторы] 

Next [<элемент>] 

где:  

элемент – переменная, которая используется для перебора всех 

элементов коллекции или массива; 

группа – имя коллекции объектов или массива. 

 Пример программы с использованием оператора For Each для 

работы с массивом (работа с коллекциями объектов будет рассмотрена 

далее): 

N=19 

ReDim x(19) 

Randomize 

For i = 0 to 19  'генерация 20-ти случайных чисел 

    x(i) = 5 - 10*Rnd 'в диапазоне от -5.000 до 4.999 

Next 

S = "" 

For Each iks in X 

    S = S & iks & vblf 

Next  

  MsgBox S,," For Each... "  

В данной программе цикл For Each использует все значения 

массива X для формирования строки S (см. рисунок 6.3). 

Задания 

С использованием оператора цикла FOR и функции RND сгенерировать 

массивы из n действительных чисел, необходимые для вычисления по 

заданной ниже формуле. Для расчета по заданной формуле использо-

вать оператор  For Each. Программу выполнить несколько раз для 

различных значений n. Исходные данные и результаты показать в окне 

сообщений. 

Лабораторная работа № 7. Процедуры и функции 
пользователя 

Процедуры и функции используются для структурирования 

сложных программ, повышающего наглядность и читаемость про-

грамм, что облегчает процессы отладки. Процедуры и функции могут 

многократно использоваться при исполнении программы, для исполне-

ния их с различными исходными данными может использоваться спи-

сок аргументов. 

Разница между процедурой и функцией:  

имя функции после ее выполнения приобретает некоторое значе-

ние (имя функции возвращает вычисленное значение), в результате она 

может использоваться в составе выражений (математических, строко-

вых и др.) программы;  

имя процедуры служит только для запуска ее в работу, результа-

том выполнения могут быть значения переменных, массивов и пр. ком-

понентов языка, являющихся аргументами процедуры или компонен-

тами, общими для вызывающей программы и процедуры. 

Синтаксис описания процедуры следующий: 

 [Public | Private] Sub <имя_проц.> [(<сп.арг.>)]  

    [<операторы>] 

    [Exit Sub] 

    [<операторы>] 

End Sub  

где: 

Public – показывает, что процедура доступна во всех других 

процедурах программы (является значением По-

умолчанию); 

Private – показывает, что процедура доступна только в тех про-

цедурах программы, где она объявлена; 

имя_проц. – имя процедуры, составленное по правилам написа-

ния идентификаторов; 

сп.арг. – список аргументов, которые передаются процедуре, 

когда она вызывается для исполнения; элементы спи-

ска разделяются запятыми; 

операторы – любая группа операторов, которая будет испол-

няться в процедуре. 

Список аргументов имеет следующий синтаксис:  

[ByVal | ByRef] <имя_переменной>[( )]  

где: 

ByVal – аргумент передается, как его значение (может быть пе-

ременной или константой при вызове процедуры, не может 

возвращать из процедуры значения); 

ByRef – аргумент передается, как ссылка; этому аргументу при 

обращении к процедуре должна соответствовать перемен-

ная (или массив) вызывающей программы и ее значение 

может изменяться после выполнения процедуры; по умол-

чанию параметры процедуры имеют тип ByRef; 

имя_переменной – имя переменной, являющейся аргументом 

процедуры, составленное по правилам написания иденти-

фикаторов.. 

Процедуры могут использовать локальные и глобальные пере-

менные. Локальные переменные могут быть объявлены в процедуре с 

использованием структуры Dim <переменные>. Переменные, которые 

используются в процедуре, но явно не объявлены, также локальные, 

если они явно не объявлены на более высоком уровне за пределами 

процедуры.  

Значения локальных переменных в процедурах не сохраняются 

между повторными вызовами процедур (области памяти под локальные 

переменные используются только во время работы процедуры).  

Процедура не может быть описана внутри другой процедуры (т. 

е. не допускается использование вложенных процедур). 

Exit Sub может использоваться для выхода из любого места 

процедуры с возвратом на следующую строку в программе после ее 

вызова.  

Вызов процедуры необходимо выполнять с использованием 

структуры [Call] <имя_проц.> [(<сп.арг.>)]. Можно не ис-

пользовать ключевое слово Call при вызове процедуры, в этом случае 

список аргументов должен быть написан без круглых скобок сразу 

именем процедуры: <имя_проц.> <сп.арг.>. Если используется 

ключевое слово Call, список аргументов должен быть написан в круг-

лых скобках: Call <имя_проц.> (<сп.арг.>).  

Пример использования процедуры в программе: 

 N = InputBox("Задайте число", _ 

    "Вычисление факториала числа") 

Call F_n(N, Fct) 

MsgBox "Факториал числа " & N & " равен " _ 

       & FCT,, "Результат расчета" 

Sub F_N(ByVal NF, FN) 

  FN=1 

  For i = 1 to NF  

     Fn = Fn * i 

  Next 

 NF = 0 'только для доказательства, что N не изменится! 
End Sub 

Результат работы программы показан на рисунке 7.1. 

  
Рисунок 7.1. Результат выполнения программы с использованием 

процедуры Sub 

Синтаксис описания функции следующий: 

[Public | Private] Function <имя_функц.>  

[(<сп.арг.>)] 

    [<операторы>] 

    [<имя_функц.> = <выражение>] 

    [Exit Function]  

    [операторы] 

    [<имя_функц.> = <выражение>] 

End Function  

где:  

имя_функц. – имя функции, составленное по правилам написа-

ния идентификаторов; 

выражение – вычисляемое значение, которое возвращает имя 

функции; 

остальные обозначения те же, что и для процедуры. 

Значение, возвращаемое функцией, по умолчанию имеет тип 

Public, если не указано иное (Private). 

Аналогично процедуре, функция не может быть вложенной в 

другую функцию. 

Пример использования функции в программе (для того, чтобы 

показать разницу в использовании процедур и функций, использован 

тот же алгоритм расчета факториала, что и в примере для процедуры): 

N = InputBox("Задайте число",_ 

   "Вычисление факториала числа") 

MsgBox "Факториал числа " & N & " равен " & F_N(N),, _ 

       "Результат расчета"  

Function F_N(NF) 

  F_N=1 

  For i = 1 to NF  

     F_N = F_N * i 

  Next 

End Function 

Результат работы программы будет тот же, что и ранее (см. рису-

нок 7.1). 

Задание 

a) Выполните задания предыдущей лабораторной работы (№ 6) с зада-

нием n в окне InputBox и использованием процедуры пользователя 

для расчета по заданной формуле. Формирование исходных данных 

и вывод результатов в окно MsgBox выполнить в головной програм-

ме. Процедура пользователя не должна использовать глобальные 

переменные. 

b) Выполните задания лабораторной работы № 5 с заданием погрешно-

сти вычисления в окне InputBox и использованием функции поль-

зователя для расчета числа π по заданной формуле. Формирование 

исходных данных и вывод результатов в окно MsgBox выполнить в 

головной программе. Функция пользователя не должна использо-

вать глобальные переменные. 

Лабораторная работа № 8. Работа с числовой 
информацией 

При выполнении математических вычислений используются 

символы математических операций: 

   = присваивание, +  сложение, - вычитание, / деление, * умноже-

ние,  ^ возведение в степень. 

Математических функция в языке VBScript достаточно много: 
Abs Atn Cbool CByte CCur CDbl CInt CLng Cos CSng CStr 

Exp Fix Int FormatCurrency FormatNumber FormatPercent 

Hex \ Lbound Log Mid Mod Oct Randomize Rnd RGB Round 

Sgn Sin Tan TypeName Ubound VarType.  

Описание функций приведено в приложении 1. 

Порядок выполнения операций в математических выражениях – 

общепринятый в математике (приоритет математических операций): 

1) математические функции, 

2) возведение в степень, 

3) умножение и деление, 

4) сложение и вычитание. 

Круглые скобки ( ) могут использоваться для изменения по-

рядка выполнения этих операций. Вначале выполняются операции в 

скобках, затем вне их. Если используются вложенные скобки, вначале 

выполняются действия во внутренних скобках, затем во внешних.  

Например, для математической записи формулы: 

 
текст в программе на языке VBScript будет следующий: 

X = (- b + (b^2 – 4*a*c)^(1/2))/(2*a) 

Особенность записи формул – нельзя пропускать знаки умноже-

ния, как это часто делается в математике. 

Для всех тригонометрических формул (Atn, Cos, Sin, Tan) 

единица измерения углов – радианы, например, число  (пи) можно 

вычислить следующим образом: 

pi = 4 * Atn(1) 

Для сложных математических формул иногда бывает целесооб-

разно выполнять вычисление по частям, например: 

 

можно вычислить следующим способом: 

y1 = (Tan(2*b + c))^2 

y2 = (a – b)^(2/3) 

y3 = (Exp(-b^2)*Log(a))/(a + Log(2*b +c)) 

y4 = abs(y2 – y3)^(1/2) 

y = y1/y4 

При записи такой формулы в одну строку получилось бы сле-

дующее выражение: 
  y = (Tan(2*b + c))^2/abs((a – b)^(2/3) – _ 
     (Exp(-b^2)*Log(a))/(a + Log(2*b +c)))^(1/2) 

которое достаточно трудно проверить на правильность написа-

ния. 

Для вычисления логарифмов с разными основаниями можно ис-

пользовать следующую программу (в примере – вычисление десятич-

ного логарифма, см. рисунок 8.1): 
n = 10 

x = 123.45 

MsgBox " Lg(" & x &") = " _ 

     & logn(x,n),,"Lg(x)" 

Function logn(xf,nf) 

  logn = Log(xf) / Log(nf) 

End Function 

Задания 

Напишите программу вычисления Y по заданному математическому 

выражению. Исходные данные и результат показать в окне сообщений. 

Вариант Выражение 
Значение переменных 

Результат 
а b c 

1)  

b

caa
bay

2
)17(

25
3/1 

  1,5 10,2 10,034 6,00 

2)  
)5(

)17/()(3
32

3

a

ca
bbay




  13,5 0,92 1,05 –20,00 

3)  
4 25

2cos

)sin(
lg c

a

ba
y 


  0,785 1,5708 3,777 5,00 

4)  
)3lg(2

3224
tg

3/2 ba

ba
y




  9,01 7,7058 – 0,05 

5)  5

13
sin2222

c

b
ay




 

2,0 13,0 0,8 2,00 

 
Рисунок 5.21. Вычисление 

логарифма Lg(x) 

6)  

cba

a

b

atg
y

2
lg

32

)(
33

3

3 3

2





  

3,1415 96,6 18,4 1,00 

7)  

)2lg(

)lg(
34

)2(tg)(tg 22

cba

a

cba
y







 
1,5 1,725 5,2425 34,00 

8)  

3/2

22

)(

)(cos)(sin
1

cba

cba

ba

y







  
3,1416 1,5708 –1,884 0,50 

9)  
c

b

acba
y

2

14
2

)sin()(lg
tg

22







 

0,5 6,385 4,201 1,00 

10)  





















cb

a
a

acb
y

2
sin)(sin

4

22

2

 9,5 –4,51 2,2093 –100,00 

Лабораторная работа № 9. Работа со строковой 
информацией 

Функции, которые могут использоваться при работе со строками, 

следующие (описание см. в Приложении 2): 
Asc Chr & InStr InStrRev Join Lcase Left Len LTrim Mid 

RTrim Trim Replace Right Space Split String StrComp 

StrConv StrReverse Tab TypeName Ucase VarType. 

Пример использования строковых функций: 

a) присвоить значение переменной  FIO 
   FIO = "Иванов Петр Сидорович" 

b) написать FIO прописными буквами 
FIO_p = Ucase(FIO) 

c) разделить  FIO  на 3 переменные: фамилию, имя и отчество (без ис-

пользования функции Split, пример с ее использованием приведен далее): 
n1 = InStr(FIO," ")  'n1= 7, позиция первого пробела в FIO 

F1 = Left(FIO,n1-1)  'F1=”Иванов”, 6 символов слева 

n2 = InStr(n1+1,fio," ",1)     'm2 = 12 

F2 = Mid(FIO, n1+1, n2-n1-1)  

             'F2 = ”Петр”, 4 (12-7-1) символа начиная с 8 

L  = Len(FIO)   'L = 21 

F3 = Mid(FIO, n2+1, L-n2) 'F3 = ”Сидорович” 

d) получить строку – инициалы и фамилия 
  F4 = Left(F2,1) & ”. ” & Left(F3,1) & ”. ” & F1 

      'F4 =”П. С. Иванов” 

e) использование функции Split: 

f = Split(FIO) 'далее можно использовать циклы 

     ' For Each или For с функциями  Lbound и Ubound 
For I = Lbound(f) to Ubound(f) 

    MsgBox f(i) 

Next 

На рисунке 9.1 показаны два примера выполнения программы, 

содержащей описанные выше операций работы со строками. 

  

Рисунок 9.1. 

Использова-

ние функций 

для обработ-

ки строковой 

информации 

Задания 

Напишите программу для своего варианта задания. Исходные 

данные и результаты показать в окне сообщений. Описание функций 

для работы со строками см. в Приложении 2. Программа должна сама 

определять количество символов или слов в исходной строке текста. 

1) В окне InputBox задать строку текста и показать код каждого сим-

вола этой строки. 

2) Показать в окне сообщений символы с кодами от 128 до 148. 

3) Определить позиции всех пробелов в строке, состоящей из любого 

числа слов. 

4) Определить позиции с конца строки всех букв «е» и «а» для строки, 

состоящей из любого числа слов. 

5) Задать текстовые значения не менее пяти элементов строкового мас-

сива и создать объединением элементов массива (Join) одну стро-

ковую переменную. 

6) Двумя вариантами (без использования и с использованием Split) 

разделить строку, состоящую из произвольного числа слов, на от-

дельные слова с присвоением полученных значений переменным. 

7) Определить количество символов в каждом слове предложения, со-

стоящего из любого числа слов. 

8) Заменить в данном предложении все буквы «е» на «Е» и «о» на «О», 

начиная с 10-го символа. 

9) Для данного предложения выполнить следующие операции: А) пре-

образовать все его символы в строчные; Б) преобразовать все сим-

волы в прописные; В) вариант «Б» преобразовать в вид исходного 

предложения. 

10) Написать все слова предложения, состоящего из любого числа 

слов, в обратном порядке (сначала последнее слово, затем предпо-

следнее и т. д., кончая первым словом). Использовать функции 

Split и Join. 

Лабораторная работа № 10. Работа с информацией 
типа дата и время 

Основной формат даты #мм/дд/гггг#, #мм-дд-гг# или #Mes-дд-

гггг, т. е. на первом месте стоит месяц, на втором – день, на третьем – 

год с разделителями косая черта с правым наклоном (/) или дефис (-). 

Например, #12/31/2008#, #12/31/8#, #12-31-2008# (31 декабря 2008 го-

да). Однако, при написании названия месяца (или 3 букв названия) мо-

жет использоваться формат, #дд-Mes-гг#, например,  #31-Dec-08#.   

При написании года двумя цифрами формат даты имеет изме-

няемый характер. Например, #12-04-08# означает 4 декабря 2008 г., но 

#13-mm-08# означает 08 <мм> 2013 г., однако #13-02-29# = 13 февраля 

2029 г.  Если на первом месте стоит число больше 12, формат преобра-

зуется в #гггг-мм-дд#. Если на последнем месте стоит число, больше 28 

или 29 для високосного года, формат преобразуется в #дд-мм-гггг#.  

Если на последнем месте стоит число, больше 30 или 31 для месяцев, в 

которых 30 и 31 день, формат преобразуется в #дд-мм-гггг#, как пока-

зано в следующем примере: 
MsgBox #13-01-31# '31.01.2013 

MsgBox #13-02-31# '13.02.1931 

MsgBox #13-03-31# '31.03.2013 

MsgBox #13-04-31#  '13.04.1931 

MsgBox #13-05-31#  '31.05.2013 

MsgBox #13-06-31#  '13.06.1931 

MsgBox #13-07-31#  '31.07.2013 

MsgBox #13-08-31#  '31.08.2013 

MsgBox #13-09-31#  '13.09.1931 

MsgBox #13-10-31#  '31.10.2013 

MsgBox #13-11-31#  '13.11.1931 

MsgBox #13-12-31#  '31.12.2013 

В связи с таким сложным поведением даты для однозначного ее 

написания лучше пользоваться четырьмя цифрами года в дате. 

Функция CDate конвертирует строку в дату с учетом региональ-

ных установок Windows. В русских настройках даты (Панель управле-

ния – Язык и региональные стандарты – текущий формат – русский) по 

умолчанию задан формат даты dd.MM.yyyy. Поэтому Cdate("13-

12-08")= 13 декабря 2008 г., тогда как #13-12-08# = 8 декабря 

2013 г.  

При неверном задании даты возникает системная ошибка (на-

пример, дата  #02/29/08# правильная – високосный год, но #02/29/07# – 

неверная  дата!).  

Функции, которые могут использоваться при работе с данными 

типа дата и время, следующие (описание см. в приложении 3): 
CDate DateAdd DateDiff DatePart DateSerial DateValue 

Day FormatDateTime Hour Minute Month MonthName Now 

Second Time TimeSerial TimeValue TypeName VarType 

Weekday WeekdayName Year. 

Дату и время на часах компьютера возвращают функции Now и 
Time. 

Если заданы два значение типа дата и время, операция вычитания 

даст разницу между ними в днях в виде действительного числа. 

Например, разница #05-02-2008 18:00# - #05-01-2008 12:00# будет 

равна 1.25 дня. 

Если необходимо вычислить разницу в определенных единицах 

(годах, кварталах, месяцах, неделях, днях, часах, минутах и секундах), 

следует использовать функцию DateDiff, как показано в следующем 

примере: 

Dt1 = #31-Dec-2005 12:00:00# 

Dt2 = #14-Apr-2009 18:01:01# 

MsgBox "Дата 1: " & Dt1 & vbLf & _ 

       "Дата 2: " & Dt2 & vbLf & _ 

 "лет (yyyy)    " & DateDiff("yyyy",dt1, dt2) & vbLf & _ 

 "кварталов (q) " & DateDiff("q", dt1,dt2) & vbLf & _ 

 "месяцев (m)   " & DateDiff("m", dt1,dt2) & vbLf & _ 

 "недель (ww)   " & DateDiff("ww",dt1,dt2) & vbLf & _ 

 "дней года (y) " & DateDiff("y", dt1,dt2) & vbLf & _ 

 "дней (d)      " & DateDiff("d", dt1,dt2) & vbLf & _ 

 "часов (h)     " & DateDiff("h", dt1,dt2) & vbLf & _ 

 "минут (n)     " & DateDiff("n", dt1,dt2) & vbLf & _ 

 "секунд (s)    " & DateDiff("s", dt1,dt2),, _ 

 "Функция DateDiff" 

Результат работы данной программы показан на рисунке 5.23.  

 

 

Рисунок 5.23. Использование 

функции DateDiff 

Аналогичные параметры задания единицы измерения имеет 

функция DateAdd, позволяющая прибавить заданный диапазон даты и 

времени к начальному значению, как показано в примере (результат 

выполнения программы – рисунок 5.24): 

Dt1 = #31-Dec-2005 12:00:00# 
MsgBox  "Дата: " & Dt1 & vbLf & vbLf & _ 
"+1 год (yyyy)    " & DateAdd("yyyy",1,dt1) & vbLf & _ 
"+1 квартал (q)   " & DateAdd("q", 1,dt1) & vbLf & _ 
"+1 месяц (m)     " & DateAdd("m", 1,dt1) & vbLf & _ 
"+1 неделя (ww)   " & DateAdd("ww",1,dt1) & vbLf & _ 
"+1 день года (y) " & DateAdd("y", 1,dt1) & vbLf & _ 
"+1 день (d)      " & DateAdd("d", 1,dt1) & vbLf & _ 
 +1 час (h)       " & DateAdd("h", 1,dt1) & vbLf & _ 
 +1 минута (n)    " & DateAdd("n", 1,dt1) & vbLf & _ 
 "+1 секунда (s)  " & DateAdd("s", 1,dt1),, _ 
 "Функция DateAdd" 

Те же параметры задания возвращаемой части даты-времени у 

функции DatePart (результат показан на рисунке 5.25): 

Dt1 = #31-Dec-2005 12:01:01# 
MsgBox  "Дата: " & Dt1 & vbLf & vbLf & _ 
 "год (yyyy)     " & DatePart("yyyy",dt1) & vbLf & _ 
 "квартал (q)    " & DatePart("q" ,dt1) & vbLf & _ 
 "месяц (m)      " & DatePart("m" ,dt1) & vbLf & _ 
 "неделя (ww)    " & DatePart("ww",dt1) & vbLf & _ 
 "день года (y)  " & DatePart("y" ,dt1) & vbLf & _ 
 "день месяца(d) " & DatePart("d" ,dt1) & vbLf & _ 
 "час (h)        " & DatePart("h" ,dt1) & vbLf & _ 
 "минута (n)     " & DatePart("n" ,dt1) & vbLf & _ 
 "секунда (s)    " & DatePart("s" ,dt1),, _ 
 "Функция DatePart" 

  
Рисунок 5.24. Использование 

функции DateAdd 

Рисунок 5.25. Использование 

функции DatePart 



49 

 

Год, месяц, день, час, секунду для заданной даты и времени мож-

но также определить с помощью функций Year, Month, Day, 

Hour, Minute, Second. 

При задании в тексте программы данных подтипа дата и время 

можно использовать английские названия месяцев (нельзя русские). 

Однако, при использовании функции преобразования строки в дату-

время все наоборот, если в региональных установках Windows задан 

русский формат дат: 
StrDt = "1 Окт 1999"  ' Строковое значение 

Data1 = CDate(StrDt)  ' Преобразование в дату: 

        ' Data1 будет равна #10-01-1999# 

Тот же результат дает функция DateValue: 
Date2 = DateValue(StrDt) 

Если заменить в региональных настройках формат дат на English, 

система перестает понимать русские названия месяцев и признает 

только английские (StrDt = "1 Oct 1999": Data1 = 

CDate(StrDt)).  

Функция Weekday(Data1, vbMonday) покажет день недели 

для заданной даты. В этой функции задан первый день недели – поне-

дельник, если опустить этот параметр, первым днем недели будет вос-

кресенье (что соответствует английскому календарю). 

Формат вывода информации подтипа дата-время можно опреде-

лить с использованием функции FormatDateTime, в которой суще-

ствует 5 форматов (vbGeneralDate, vbLongDate, vbShortDate, 

vbLongTime, vbShortTime, описание этих и других констант – в 

Приложении 4). 

Название 7-го дня недели позволяет определить функция 

WeekDayName(7,False,vbUseSystem) – в данном случае при рус-

ских региональных настройках операционной системы вернет «воскре-

сенье». 

Задания 

Для приведенных ниже вариантов заданий исходные данные и резуль-

таты показать в окне сообщений.   

1) Задайте в программе строковую переменную, значение которой рав-

но текущей дате с написанием в ней месяца названием. Преобразуй-

те значение переменной в подтип «дата». Вычислите количество 

прожитых Вами дней.  

2) Определите текущую дату на часах компьютера, прибавьте к ней 1 

год, затем 3 месяца и 25 дней и определите название дня недели по-

лученной даты. 

3) Задайте в программе строковую переменную, значение которой рав-

но текущему времени с точностью секунд. Преобразуйте значение 

переменной в подтип «время». Вычислите количество секунд, ос-

тавшихся до конца суток.  

4) Определите текущую дату и время на часах компьютера, прибавьте 

к нему 25 часов, 30 минут и 30 секунд, и определите для полученно-

го значения количество часов, минут и секунд, прошедших от нача-

ла суток. 

5) Вычислите количество дней, часов, минут и секунд, прошедших с 

начала 21 века до текущего момента, который взять с часов компью-

тера. 

6) Рассчитайте стаж работника – количество целых лет, кроме того це-

лых месяцев и дней (например, 10 лет 1 месяц и 1 день) к текущему 

моменту времени, который определить по часам компьютера.  

7) Рассчитайте количество рабочих дней при пятидневной рабочей не-

деле с 1.09.2010 по 30.11.2010. 

8) Рассчитайте количество выходных дней при пятидневной и шести-

дневной рабочей неделе с 1.01.2010 по 31.10.2010. 

9) Рассчитайте количество отработанных часов за период с 1.03.2010 

по 31.05.2010 с учетом того, что в этом периоде один праздничный 

день. 

10) Рассчитайте количество учебных часов за период с 1.09.2008 по 

31.11.2010 с учетом того, что количество их по дням недели сле-

дующее: понедельник – 6, вторник – 4, среда – 8, четверг – 5, пятни-

ца – 4, суббота и воскресенье – нет занятий.   

Лабораторная работа № 11. Работа с логическими 
выражениями 

При написании логических условий используются операции 

сравнения данных (подтипов число, строка, дата и время): 

= равно,  

<> не равно,  

< меньше,  

<= меньше или равно,  

> больше,  

>= больше или равно. 

Кроме того, используются логические операции: 
And Eqv Imp Is IsArray IsDate IsEmpty IsNull IsNumeric 

Not Or Xor. 

Результатом выполнения логической операции является одно из 

двух возможных значений:  

True (Истина) или  

False (Ложь). 

Переменной можно присвоить логическое значение True или 

False (L1 = True или L1 = False), но нельзя использовать русские зна-

чения Истина или Ложь в присвоении и логических выражениях, не 

смотря на то, что русские версии Windows в окне сообщений показы-

вают именно эти значения (см. рисунок 11.1).  

При сравнении символьных и строковых значений учитывается 

регистр букв (прописные или строчные).  

Приоритет при вычислении выражений, в которых присутствуют 

логические компоненты: 

1) арифметические операторы; 

2) операторы объединения (конкатенации) строковых значений (&, +); 

3) операции сравнения данных, которые используют символы =, <>, <, 

<=, >, >=; 

4) логические операции And, Or, Not, Xor, Eqv, Imp. 

В операциях одного приоритета порядок вычислений слева-

направо. Порядок может быть изменен при использовании круглых 

скобок. 

Результат вычисления логического выражения может быть при-

своен переменной, показан в окне MsgBox, использован для организа-

ции разветвляющихся алгоритмов в операторах If…., Select 

Case… и циклов в структурах Do [While | Until]…, While. 

Пример программы с использованием логических выражений: 

    a = 1 > 10    'False 

    b = "я" > "Я"    'True 

    c = "Иванов" < "Петров"     'True 

    d = 1 < 10 and "я" > "Я"  'True 

    e = 1 > 10 or #31-12-2007# < #01-01-2008# 'True 
    MsgBox a & vbLF & b & vbLF &c & vbLF & d & vbLF & e 

Результат выполнения приведен-

ного выше текста программы показан 

на рисунке 11.1. 

Краткое описание логических 

операций (здесь и в таблице 5.5 E1 и 

E2 – логические выражения.): 

 And – логическое И (если выражения 

слева и справа от него истинны, ре-

зультат True, иначе False или 

Null); 

 Or – логическое ИЛИ (должно быть 

истинным хотя бы одно из вы-

ражений); 

 Not – логическое отрицание (возвращает True, если условие ложно 

и наоборот); 

 Xor – логическое исключение (выражение E1 Xor E2 возвращает 

True, если только E1 = True или только E2 = True, иначе – 

False); 

 Eqv – эквивалентность двух выражений, возвращает True, если они 

имеют одинаковое значение; 

 Imp – импликация (E1 Imp E2 возвращает False, если E1 = True 

и E2 = False, иначе – True). 

Полное описание результатов вычисления логических операций 

приведено в таблице 5.5. 

Таблица 5.5. Результаты логических операций 
E1 E2  E1  And  E2  E1 Or  E2  E1 Imp  E2  E1  Xor  E2 

True True True True True False 

True False False True False True 

True Null Null True Null  

False True False True True True 

False False False False True False 

False Null False Null True  

Null True Null True True  

Null False False Null Null  

Null Null Null Null Null  

Задания 

В окне сообщений показать результаты вычисления логических значе-

ний: 

1) "A" > "f" и 1e-10=1/1e10 

  

Рисунок 11.1.  Результат 
вычисления логических 

функций в русской и 
английской версиях Windows 

2) «Правда» > «Ложь» или True не равно False 

3) Sin(0,5) > cos(0,5) и «Саша» > «Леши» 

4) «эврика» = «eureka» или «привет!» = «Hi!» 
5) 0 часов 5 мин. 1.1.2008 > 20 часов 15 мин. 31.12.2007 

6) vbOKOnly > vbOKCancel и vbCritical < vbQuestion 

7) vbLf=Chr(10) и vbNullChar = Chr(0) 

8) vbSunday = 1 и vbMonday=2 и vbTuesday=3 

9) vbBlack=&h00 и vbRed=&hFF и vbWhite = &hFFFFFF 

10) Exp(1) = Exp(-1) или Abs(Sin(1)) = Abs(Sin(-1)) 

Лабораторная работа № 12. Работа с объектами 
WScript 

 Программа на языке VBScript сама по 

себе является объектом WScript сервера 

сценариев Windows Script Host, (т. е. после 

запуска программы этот объект уже сущест-

вует, не нужно давать команду о его созда-

нии). Объект WScript – корневой в объект-

ной иерархии Windows Script Host, в которую 

входят три COM-библиотеки (COM-

Component Object Model): WshController, 

WshNetwork и WshShell (см. рисунок 12.1 с 

сайта http://msdn.microsoft.com/en-us/libra-

ry/a74hyyw0(VS.85).aspx).  

В программах на VBScript могут ис-

пользоваться и другие библиотеки классов 

Windows (например, объекты библиотеки 

Microsoft ADO – классы для работы с ба-

зами данных и пр.), библиотеки других сис-

тем, поддерживающих интерфейс ActiveX 

(OLE Automation), например, библиотеки Mi-

crosoft Offfice, Internet Explorer и пр.  

Библиотеки классов содержат описания объектов, образующих 

иерархическую структуру.  

Для создания в программе экземпляра объекта (не Wscript!) ис-

пользуется следующий синтаксис оператора присваивания: 

Set <Переменная> = CreateObject(“<Библиотека.Класс>”) 

C использованием этой функции создается переменная подтипа 

Object, после чего можно узнать свойства созданного объекта и поль-

зоваться его методами (процедурами и функциями объекта). 

Свойство объекта имеет определенное значение, которое можно 

узнать или изменить. 

Значение свойства объекта можно использовать в операторе при-

сваивания и в любых других выражениях с использованием синтаксиса 

<Имя объекта>.<Имя свойства> (имя объекта и имя его свойства, 

разделенные точкой), например: 

<Переменная> =  <Имя объекта>.<Имя свойства> 

или 
MsgBox <Имя объекта>.<Имя свойства> 

 

Рисунок 12.1. 

Объектная модель 

Windows Script Host 

В некоторых случаях свойство может возвращать ссылку на кол-

лекцию объектов или значений, в этом случае возможно только при-

своение этого свойства переменной с использованием слова Set, как 

при создании объекта: 

Set <Переменная> = <Имя объекта>.<Имя метода> 

Чтобы изменить свойство объекта, достаточно присвоить ему ка-

кое-либо значение: 

<Имя объекта>.<Имя свойства> = <Значение> 

Присваиваемое значение может быть константой, выражением, 

свойством другого объекта, возвращаемым значением какого либо ме-

тода: 

 <Имя объекта>.<Имя свойства> =  _ 

            <Имя объекта2>.<Имя метода([параметры])> 

Метод объекта – это его процедура, которая может  выполнять 

какие-либо действия, получать и возвращать значения (параметры ме-

тода).  

При использовании метода с параметрами следует использовать 

следующий синтаксис: 

1) если метод не возвращает значений: 

<Имя объекта>.<Имя метода> <п1> [,<п2>, …]> 

где  п1, п2 и т. д. – параметры; 

2) если метод возвращает какое-либо значение (но не объект и не 

коллекцию объектов), обязательно использование круглых 

скобок: 

<Переменная> = <Имя объекта>.<Имя метода>([<п1> [,<п2>, …]]>) 

3) если метод возвращает ссылку на новый объект или их кол-

лекцию – обязательно использование слова Set, как при соз-

дании объекта: 

Set <Переменная> = <Имя объекта>.<Имя метода> <п1> _ 

                   [,<п2>, …]> 

При работе с объектами может использоваться структура языка 

VBScript: 
With <Имя объекта> 

      операторы 

End With 

Для операторов между With и End With можно использовать 

методы и свойства заданного объекта, которые начинаются с точки (.) 

без указания предшествующего имени объекта, как показано далее в 

примерах программ. 

Объект WScript имеет следующие методы: Sleep Quit Echo 

CreateObject ConnectObject DisconnectObject GetObject,  

и свойства: Arguments BuildVersion FullName Interac-
tive Name Path ScriptFullName ScriptName Timeout Ver-

sion StdIn StdOut StdErr (последние 3 свойства можно использо-

вать только для программы, запущенной с помощью CScript.exe, т. е. в 

окне Cmd).  

Объект Wscript имеет подчиненный объект Shell, с методами: 
AppActivate  CreateShortcut Exec ExpandEnviron-

mentStrings LogEvent Popup Run и пр., 

и свойствами: CurrentDirectory Environment Special-

Folders. 

 Описание этих объектов, их методов и свойств можно найти в 

справке системы Microsoft Visual Studio или на сайте корпорации Mi-

crosoft. 

Методы объекта могут имеют один параметр (например, Exec 

<Command>) или несколько параметров. Данный метод используется 

для запуска исполняемых файлов. 

Так, метод Run имеет следующий синтаксис: 

Run(<Command>,<WindowStyle>,<WaitOnReturn>) 

где   Command – командная строка запуска приложения; 

WindowStyle – необязательный параметр, число, опреде-

ляет вид окна запускаемого приложения: 

1 – запускает программу в окне обычного размера; 

2 – запускает программу в свѐрнутом окне; 

3 – запускает программу в максимизированном окне; 

 WaitOnReturn - необязательный параметр, логическое 

значение: C – программа VBS возобновит работу только 

после завершения вызванного процесса, False (по умолча-

нию) – управление передаѐтся сразу на следующую строку 

программы. 

Метод Run позволяет в качестве параметра Command указывать 

имя файла данных, известного для Windows типа. Путь к файлу данных 

следует указывать в формате 8.3 (без пробелов и длинных названий па-

пок). 

Методы объекта WScript позволяют не только управлять вы-

полнением программы, но и устанавливать связь с другими объектами 

ActiveX (OLE Automation) –  Wscript.ConnectObject(<ProgID>, 

<Prefix>), создавать объекты существующих библиотек – 

Wscript.CreateObject(<ProgID>, <Prefix>), а также создавать 

объект из его описания в файле – Wscript.GetObject(<Pathname>, 

<ProgID>, <Prefix>).  

Свойства данного объекта позволяют узнать параметры запущен-

ной программы,  кроме того свойством Wscript.Timeout можно за-

дать максимальную продолжительность работы программы (в милли-

секундах), а заданием значения свойства WScript.Interactive = 

False  можно запретить вывод программой диалоговых окон.  

 Пример программы, сообщающей все свойства о самой себе (ре-

зультат выполнения программы показан на рисунок 12.2): 

With WScript 

  .echo(.BuildVersion & vbLF & .FullName & vbLF & _ 

        .Interactive & vbLF  &  .Name   & vbLF & _ 

        .Path & vbLF & .ScriptFullName & vbLF & _ 

        .ScriptName & vbLF & .Timeout & vbLF &  _ 

     .Version) 

End With 

Пример программы, использующей метод WScript.Sleep, чтобы 

остановить работу программы на заданное количество миллисекунд 

(результат выполнения программы показан на рисунок 12.3): 

t1 = time 

WScript.Sleep 25545 

t2 = time 

h1 = Hour(t1): m1 = Minute(t1) : s1 = Second(t1) 

h2 = Hour(t2): m2 = Minute(t2) : s2 = Second(t2) 

ds = (h2-h1)*3600 + (m2-m1)*60 + (s2-s1) 

MsgBox("Метод WScript.Sleep 25545" & vbLF & vbLF &_ 

  "Tнач = " & t1 & " Tкон = " & t2 & vbLF & vbLF &_ 

  "Программа проспала " & ds & " секунд"), _ 

   vbExclamation, "Использование метода WScript.Sleep" 

  

Рисунок 12.2. Свойства 

объекта WScript 

Рисунок 12.3. Использование 

метода WScript.Sleep 

 Использование метода Popup объекта WScript.Shell было 

рассмотрено ранее в лабораторной работе №1.  Аналогично следует 

использовать методы AppActivate, Exec, Run и пр. 

Задания 

1) С использованием метода Exec объекта WScript.Shell запусти-

те из своей программы Калькулятор Windows. 

2)  С использованием метода Exec объекта WScript.Shell запусти-

те из своей программы Блокнот Windows. 

3) С использованием метода Exec объекта WScript.Shell запусти-

те из своей программы Explorer Windows. 

4) С использованием метода Exec объекта WScript.Shell запусти-

те из своей программы Internet Explorer Windows. 

5) С использованием метода Exec объекта WScript.Shell запусти-

те из своей программы приложение  Microsoft Office Word. 

6) С использованием метода Run объекта WScript.Shell запустите 

из своей программы Блокнот Windows в свернутом окне 

7) С использованием метода Run объекта WScript.Shell запустите 

из своей программы Wordpad Windows в окне, развернутом на весь 

экран. 

8)  Напишите программу, которая будет запускать Блокнот Windows  

и покажет после закрытия Блокнота сообщение о продолжительно-

сти его работы в минутах и секундах. 

9) Напишите программу, которая с использованием метода AppActi-

vate(<Имя окна приложения>) объекта WScript.Shell опре-

делит, запущена или нет в системе программа Блокнот   

10) Напишите программу, которая с использованием метода AppAc-

tivate(<Имя окна приложения>) объекта WScript.Shell оп-

ределит, запущена или нет в системе программа Калькулятор и, ес-

ли программа не запущена, запустит ее.  

Лабораторная работа № 13. Работа с информацией 
файловой системы 

 Система Microsoft Visual Basic 

Scripting Edition для создания объектов, 

работающих с файловой системой, исполь-

зует библиотеку классов Windows с име-

нем Scripting (Microsoft Scripting 

Runtime Library – файл  

…\windows\system32\Scrrun.dll).  

Для дальнейшего понимания назна-

чения объектов, их свойств и методов не-

обходимо знание английской компьютер-

ной терминологии. 

Список объектов, которые могут 

быть созданы при работе с этой библиоте-

кой, показан на рисунке 5.31. Полное опи-

сание объектов можно найти, например, в 

файле 
…\Microsoft Office\Office12\1049\Vbscrip5.сhm. 

Главный объект библиотеки классов 

Scripting –  

Scripting.FileSystemObject, который 

имеет вложенные объекты Drive, Folder, File и TextStream.  

Методы объекта FileSystemObject: BuildPath CopyFile 

CopyFolder CreateFolder CreateTextFile DeleteFile Dele-

teFolder DriveExists FileExists FolderExists GetAbsolu-

tePathName GetBaseName GetDrive GetDriveName GetExten-

sionName GetFile GetFileName GetFolder GetParentFolder-

Name GetSpecialFolder GetTempName MoveFile MoveFolder 

OpenTextFile VarType. Объект имеет одно свойство – Drives,  

возвращающее ссылку на коллекцию объектов Drive – дисковых уст-

ройств в системе.  

Коллекция – упорядоченное множество однотипных объектов. 

Любая коллекция имеет свойство Count (количество объектов в 

коллекции). 

Ссылки на коллекции возвращает также метод GetFolder: на 

коллекцию SubFolders (объекты Folder) и на коллекцию Files 

(объекты File).  

Свойства объекта Drive: AvailableSpace DriveLetter 

DriveType FileSystem FreeSpace IsReady Path RootFolder 

 
Рисунок 13.1.  Библиотека 

классов Scripting 

file:///C:\ASG\�������\Informatika\���_������_�����������\����������������\VBScript\Microsoft%20VBScripr%20Reference\vsproAvailableSpace.htm
file:///C:\ASG\�������\Informatika\���_������_�����������\����������������\VBScript\Microsoft%20VBScripr%20Reference\vsproAvailableSpace.htm
file:///C:\ASG\�������\Informatika\���_������_�����������\����������������\VBScript\Microsoft%20VBScripr%20Reference\vsproAvailableSpace.htm
file:///C:\ASG\�������\Informatika\���_������_�����������\����������������\VBScript\Microsoft%20VBScripr%20Reference\vsproDriveType.htm
file:///C:\ASG\�������\Informatika\���_������_�����������\����������������\VBScript\Microsoft%20VBScripr%20Reference\vsproDriveType.htm
file:///C:\ASG\�������\Informatika\���_������_�����������\����������������\VBScript\Microsoft%20VBScripr%20Reference\vsproDriveType.htm
file:///C:\ASG\�������\Informatika\���_������_�����������\����������������\VBScript\Microsoft%20VBScripr%20Reference\vsproFreeSpace.htm
file:///C:\ASG\�������\Informatika\���_������_�����������\����������������\VBScript\Microsoft%20VBScripr%20Reference\vsproFreeSpace.htm
file:///C:\ASG\�������\Informatika\���_������_�����������\����������������\VBScript\Microsoft%20VBScripr%20Reference\vsproFreeSpace.htm
file:///C:\ASG\�������\Informatika\���_������_�����������\����������������\VBScript\Microsoft%20VBScripr%20Reference\vsproPath.htm
file:///C:\ASG\�������\Informatika\���_������_�����������\����������������\VBScript\Microsoft%20VBScripr%20Reference\vsproPath.htm
file:///C:\ASG\�������\Informatika\���_������_�����������\����������������\VBScript\Microsoft%20VBScripr%20Reference\vsproPath.htm

SerialNumber ShareName TotalSize VolumeName.  Методов 

объект не имеет. 

Методы объектов Folder и File: Copy Delete Move Ope-
nAsTextStream.  

Свойства объектов Folder и File: Attributes DateCreated 
DateLastAccessed DateLastModified Drive Name Parent-

Folder Path ShortName ShortPath Size Type.  

Методы объекта TextStream: Close Read ReadAll Read-
Line Skip SkipLine Write WriteLine WriteBlankLines.  

Свойства объекта TextStream: AtEndOfLine AtEndOfStream 
Column Line.  

В библиотеке классов Scripting присутствует объект Dictionary 

(словарь), который предназначен для сохранения множества парных 

значений, первый элемент каждой пары – ключевое (уникальное) зна-

чение. Этот объект имеет свойства: CompareMode, Count, Item, 

Key и методы: Add, Exists, Items, Keys, Remove, RemoveAll. 

Некоторые примеры использования объектов приведены ниже.  

Пример 1. Информация о дисках компьютера  (результат испол-

нения – рисунок 13.2). 

  On Error Resume Next  'на случай, если в CD-приводе нет диска 
  Dim  fs, dr, ndr, d, d1, drv, s 

  Set fs = CreateObject("Scripting.FileSystemObject") 

  Set dr = fs.Drives 

  ndr = dr.Count 

  d = "Всего дисков - "  & ndr & ": " 

  For Each d1 in dr 

    name_d = d1.DriveLetter & ":"  

    d = d & name_d  & " " 

    Set drv = fs.GetDrive(fs.GetDriveName(name_d)) 

    s = s & vbLF & "Имя диска " & drv.Path & " - "  

    s = s & drv.VolumeName   

    Select Case drv.DriveType 

     Case 0: t = "Неизвестный" 

     Case 1: t = "Удаляемый (флэш и пр.)" 

     Case 2: t = "Раздел винчестера" 

     Case 3: t = "Сетевой" 

     Case 4: t = "CD-ROM" 

     Case 5: t = "RAM диск" 

   End Select 

    s = s & ", тип " & t & vbLF    

    s = s & "Серийный номер диска " & _ 

        drv.SerialNumber & vbLF 

    s = s & "Всего: " & _ 

        FormatNumber(drv.TotalSize/1024/1024, 0)  

    s = s & " Мб,"  

    s = s & " свободно: " & _ 

        FormatNumber(drv.FreeSpace/1024/1024, 0)  

    s = s & " Мб" & vbLf 

   Next 

  s = d & vbLf & s 

  MsgBox s,, "Это все FileSystemObject!"   

Пример 2. Информация о папках  (результат исполнения – рису-

нок 13.3). 
Dim fso, f, f1, fc, s 

Set fso = CreateObject("Scripting.FileSystemObject") 

s = "C:\inetpub" 

Set f = fso.GetFolder(s) 

Set fc = f.SubFolders  'коллекция объектов SubFolders 

s = s & ", папки:    " & vbLF 

For Each f1 in fc     

  s = s & f1.name & vbLF      'папки в коллекции SubFolders 

Next 

MsgBox s,, "Коллекция SubFolders" 

 

 

Рисунок 13.3. Пример 2 

 

Рисунок 13.2. Пример 1 Рисунок 13.4. Пример 3 

Пример 3. Информация о файлах и их типах в папке  (результат 

исполнения – рисунок 13.4). 
Dim fso, f, f1, fc, s 

Set fso = CreateObject("Scripting.FileSystemObject") 

Set f = fso.GetFolder("C:\inetpub\wwwroot") 

Set fc = f.Files  'коллекция объектов Files 

For Each f1 in fc 

      s = s & f1.name & "  тип: " & f1.Type & vbLF  

Next 

MsgBox s,, "Коллекция Files" 

  Пример 4. Чтение информации о файле (результат исполнения – ри-

сунок 13.5). 
Dim  fso, fld, F_name, f, s  

Set fso = CreateObject("Scripting.FileSystemObject") 

F_name = "D:\ASG\Мой учебник по " &_ 

  "информатике\VBScript\Чтение информации о файле.vbs"  

Set f = fso.GetFile(F_name)    'Создан объект File 
s = "Диск - " & f.Drive.DriveLetter & vbLf 

s = s & "Путь - " & f.Path & vbLf 

s = s & "Путь в формате 8.3 - " & f.ShortPath & vbLf 

s = s & "Имя - " & f.Name & vbLf 

s = s & "Короткое имя - " & f.ShortName & vbLf 

s = s & "Размер - " & f.Size & " байт" & vbLf 

s = s & "Тип файла - " & f.Type & vbLf 

s = s & "Дата создания - " & f.DateCreated & vbLf 

s = s & "Дата последнего доступа - " & _ 

f.DateLastAccessed & vbLf 

s = s & "Дата последней модификации - " & _ 

  f.DateLastModified 

MsgBox s,, "Свойства файла" 

Пример 5. Чтение строк из файла (результат исполнения – рису-

нок 13.6). 

Dim  fso, F_name, f_txt, s  

Set fso = CreateObject("Scripting.FileSystemObject") 

F_name = "D:\VBScript\Prg_obj5.vbs" 

Set f_txt = fso.OpenTextFile(F_name, 1) 

Do While f_txt.AtEndOfStream <> True 

    s = s + f_txt.ReadLine + vbLF 

Loop 

f_txt.Close 

MsgBox S,, "Чтение строк из файла" 

  

Рисунок 13.5. Пример 4 Рисунок 13.6. Пример 5 

Пример 6. Создание папки, текстового файла и запись в файл: 

Dim  fso, fld, F_name, f_txt, ns, s 

s = "" 

Set fso = CreateObject("Scripting.FileSystemObject") 

fld = "C:\vbscripts\" 

If Not (fso.FolderExists(fld)) Then  

   fso.CreateFolder(fld) 

   s = "Cоздана папка " & fld & "!" & vbLF 

End If 

F_name = fld & "vbs_test1.txt" 

If Not (fso.FileExists(F_name)) Then 

   Set f_txt = fso.CreateTextFile(F_name, True) 

   f_txt.WriteLine("1. Файл " & F_name & " создан!") 

   s = s & "Cоздан файл " & F_name & "!" & vbLF 

   s = s & " о чем записано в 1-ой строке файла!" & 

vbLF 

End If  

Set f_txt = fso.OpenTextFile(F_name, 1) 

            'Файл открыт для чтения, как объект TextStreem 
f_txt.ReadAll 

ns = f_txt.Line 'Номер последней строки минус 1 
Set f_txt = fso.OpenTextFile(F_name,8)  

                'Открыт для дозаписи в конец файла 
f_txt.WriteLine(ns & "-я строка текстового файла.") 

f_txt.Close 

s = s & "В конец файла " & F_name & vbLF   

s = s & "  добавлена " & ns & "-я строка" 

s = s & vbLF & vbLF & "Всего строк в файле - " & ns 

MsgBox s,, "Добавление строк в файл" 

 После запуска на исполнение этой программы первый раз (при 

отсутствии папки и файла) будет пока-

зано сообщение, приведенное на ри-

сунке 13.7. После запуска на исполне-

ние программы 6 три раза файл  

vbs_test1.txt будет содержать 

текст: 

1. Файл C:\vbscripts\vbs_test1.txt создан! 
2-я строка текстового файла. 
3-я строка текстового файла. 
4-я строка текстового файла.  

Рисунок 13.7. Пример 6 

Пример 7. Генерирация случайного массива из 10 млн. чисел 

(или больше…), запись чисел в текстовый файл, создаваемый програм-

мой, и открытие этого файла в Блокноте Windows: 

Set FSO = CreateObject("Scripting.FileSystemObject") 

Set Txt = FSO.CreateTextFile("D:\Test.txt") 

n=10000000 : a=-10 : b=10  

ReDim x(n) 

Randomize 

t1=Time : L = "" : xsr=0 

For i=1 To n 

  x(i) = b - (b-a)*Rnd 

  xsr = xsr + x(i) 

  L = L & FormatNumber(x(i),2) & vbTab 

  If i/10 = i\10 Then Txt.WriteLine(L) : L = "" 

Next 

xmin=x(1): xmax=x(1) : xsr = xsr/n 

For i=2 To n 

  If x(i)<xmin Then xmin=x(i) 

  If x(i)>xmax Then xmax=x(i) 

Next 

t2=Time 

Txt.Write(" Минимальное  X = " & xmin & vbNewLine) 

Txt.Write(" Максимальное X = " & xmax & vbNewLine) 

Txt.Write(" Среднее X = " & xsr & vbNewLine) 

Txt.Write(" Время работы программы = " & 

FormatDateTime(t2-t1)) 

Txt.Close 

Set rf = CreateObject("Wscript.Shell") 

rf.Run("notepad D:\Test.txt") 

В текстовый файл NTFS теоретически можно записать несколько 

Тбайт информации, в то время, как в окне сообщений MsgBox длина 

строки сообщений ограничена 256 символами, поэтому в нем более 50 

чисел показать не удастся.  

В программах на VBScript можно использовать вывод результа-

тов работы программы в текстовый файл вместо окна MsgBox. 

Конец файла Test.txt, открытого в Блокноте (последняя строка 

имеет номер 1000000, в строке по 10 чисел), показан на рисунке 13.8. 

 
Рисунок 13.8. 10 млн. чисел, записанных программой в файл 

Пример 8. Копирование файла: 

Const OverwriteExisting = TRUE 

Set fso = CreateObject("Scripting.FileSystemObject") 

fso.CopyFile "C:\vbscripts\vbs_test1.txt", _ 

          "C:\vbscripts\vbs_test2.txt ", OverwriteExisting 

Результатом работы программы будет новый файл 

vbs_test2.txt. Если запустить программу повторно, старый файл 

vbs_test2.txt будет заменен новым (т. к. параметр OverwriteEx-

isting = TRUE, если задать OverwriteExisting =FALSE появится 

сообщение об ошибке «File already exist» –  «Файл уже существует»). 

Пример 9. Использование объекта Dictionary для чтения атрибу-

тов файла (результаты работы – см. рисунок 13.9): 

Set FSO = CreateObject("Scripting.FileSystemObject") 

Set File = FSO.GetFile("C:\bootmgr") 

Attrs = File.Attributes 

Set Dict = CreateObject("Scripting.Dictionary") 

Dict.Add "Обычный", "Нет" 

Dict.Add "Только чтение", "Нет" 

Dict.Add "Скрытый", "Нет" 

 Dict.Add "Системный", "Нет" 

Dict.Add "Том диска", "Нет" 

Dict.Add "Папка", "Нет" 

Dict.Add "Архивный", "Нет" 

Dict.Add "Ярлык", "Нет" 

Dict.Add "Сжатый", "Нет" 

If Attrs=0 Then 

    Dict.Item("Обычный") = "Да" 

End If 

If Attrs>=2048 Then 

    Dict.Item("Сжатый") = "Да" 

    Attrs = Attrs-2048 

End If 

If Attrs>=1024 Then 

    Dict.Item("Ярлык") = "Да" 

    Attrs = Attrs-1024 

End If 

If Attrs>=32 Then 

   Dict.Item("Архивный") = "Да" 

   Attrs = Attrs-32 

End If 

If Attrs>=16 Then 

    Dict.Item("Папка") = "Да" 

    Attrs = Attrs-16 

End If 

If Attrs>=8 Then 

 

Рисунок 13.9. Пример 8 

    Dict.Item("Том диска") = "Да" 

    Attrs = Attrs-8 

End If 

If Attrs>=4 Then 

    Dict.Item("Системный") = "Да" 

    Attrs = Attrs-4 

End If 

If Attrs>=2 Then 

    Dict.Item("Скрытый") = "Да" 

    Attrs = Attrs-2 

End If 

If Attrs>=1 Then 

    Dict.Item("Только чтение") = "Да" 

    Attrs = Attrs-1 

End If 

Str = "Атрибуты файла """ & File.Path & """:" & vbCrLf 

For Each Attr In Dict 

    Str = Str & Attr & " = " & Dict.Item(Attr) & 

vbCrLf 

Next 

MsgBox Str 

Задания 

1) Напишите программу, которая создаст в Вашей папке новый тексто-

вый файл и пронумерует в нем строки с 1 по 50, затем создаст ко-

пию этого файла (с другим именем) в той же папке. 

2) Напишите программу, которая прочитает текст из 5-ти файлов Ва-

ших VBS-программ и запишет весь этот текст в новый файл. В но-

вом файле перед текстом каждой программы написать название 

файла (перед названием вставить пустую строку). Откройте этот 

файл методом Run в программе по умолчанию для текстовых фай-

лов. 

3) Напишите программу, которая прочитает информацию о файлах  

Ваших программ в Вашей папке: имя файла, имя и путь, имя и путь 

в формате 8.3, дата создания, дата последнего изменения, размер 

файла и запишет эту информацию в новый текстовый файл и откро-

ет его методом Run в программе по умолчанию для текстовых фай-

лов. 

4) Напишите программу, которая прочитает информацию о дисковых 

устройствах на Вашем компьютере: общее количество, имя, тип, 

общий объем и свободный объем каждого диска, запишет эту ин-

формацию в новый текстовый файл и откроет его методом Run в 

программе по умолчанию для текстовых файлов. 

5) Напишите программу, которая откроет файл Vbscrip5.сhm или анало-

гичный в предыдущих версиях Office. Предварительно найдите этот 

файл, скопируйте путь к нему и название в программу. Если файл не 

открывается методом Exec объекта WScript.Shell, попробуйте 

сделать это методом Run. Если файл не открывается, воспользуй-

тесь свойством  ShortPath объекта File библиотеки 

Scripting.FileSystemObject для преобразования имени и пути 

в формат 8.3.  Покажите в окне InputBox преобразованное имя и 

путь. Все варианты задания сохранить, сообщения системы привес-

ти в отчете. 

6) Аналогично заданию 5 напишите программу открытия файла графи-

ческого формата (*.bmp, *.jpg, *.png или пр.) из папки с именем 

«Мои новые рисунки» в папке документов своего профиля пользо-

вателя. 

7) Напишите программу, которая создаст новую папку «Мои новые 

файлы» в папке документов Вашего профиля пользователя, в новой 

папке создаст текстовый файл с помощью метода CreateTextFile 

библиотеки Scripting.FileSystemObject и откроет его методом 

Run в программе по умолчанию для текстовых файлов.  

8) Напишите программу, которая создаст текстовый файл в папке до-

кументов Вашего профиля пользователя и запишет в него информа-

цию о той папке, в которой находится данный файл (имя папки, имя 

в формате 8.3, дату создания, путь к папке, короткий путь в формате 

8.3, размер папки и ее атрибуты. 

9) Напишите программу, которая  создаст текстовый файл в папке до-

кументов Вашего профиля пользователя и запишет в него информа-

цию о всех папках в папке C:\ на Вашем ПК и об их атрибутах. 

10) Напишите программу, которая покажет в окне сообщений все 

файлы с расширением exe в папке C:\Windows. 

Лабораторная работа № 14. Работа с информацией 
об ошибках 

Если написать в начале программы структуру 

On Error Resume Next 

автоматическое завершение работы программы при возникновении 

ошибки будет предотвращено и появится возможность выполнить не-

которые дополнительные действия в программе с использованием объ-

екта  Err языка VBScript, который автоматически создается при рабо-

те программы и значение свойства Number которого в этом случае ста-

нет не равным нулю, что можно использовать для программной обра-

ботки ошибок. 

Методы объекта Err: Clear Raise. Свойства объекта Err: 
Description HelpContext HelpFile Number. 

Пример программы с обработкой ошибки (дополненный пример 

7 предыдущей лабораторной работы): 

On Error Resume Next 

Const OverwriteExisting = False 

Set fso = CreateObject("Scripting.FileSystemObject") 

fso.CopyFile "C:\vbscripts\vbs_test1.txt", _ 

       "C:\vbscripts\vbs_test2.txt ", _ 

        OverwriteExisting 

IF Err.Number <> 0 Then 'файл уже существует 
   MsgBox "Файл уже существует и не может быть" & _        

       "заменен новым", vbCritical,"Сообщение об 

ошибке" 

End If 

Пример использования для Примера 5 (см. выше): 

On Error Resume Next 

Dim  fso, F_name, f_txt, s  

Set fso = CreateObject("Scripting.FileSystemObject") 

F_name = "C:\vbscripts\vbs_test1.txt"  

Set f_txt = fso.OpenTextFile(F_name, 1) 

If Err.Number = 0 Then   'файл существует и открыт 

       'далее –  цикл чтения строк из файла 
  Do While f_txt.AtEndOfStream <> True 

    s = s + f_txt.ReadLine + vbLF 

  Loop 

  f_txt.Close 

  MsgBox S,, "Чтение строк из файла" 

Else 

 MsgBox ("Ошибка № " & CStr(Err.Number) & vblf  _ 

         & Err.Description & vblf & Err.Source) 

 Err.Clear    ' очистка свойств объекта Err 

file:///C:\ASG\�������\Informatika\���_������_�����������\����������������\VBScript\Microsoft%20VBScripr%20Reference\vsproSource.htm

End If   

Впрочем, данная ошибка может быть выявлена и другим спосо-

бом: … If IsObject(f_txt) Then 'объект создан … 

Задание 

Программу, написанную Вами ранее в лабораторной работе 12 

или 13, в которой выполняется открытие существующего файла, до-

полните фрагментом обработки ошибки открытия в случае неверного 

указания пути или имени файла с выводом сообщения об этом.    

Лабораторная работа № 15. Работа с информацией 
локальной сети 

Для этой цели может использоваться объект WScript.Network, 

имеющий следующие свойства: ComputerName UserName UserDo-

main и методы: EnumNetworkDrives EnumPrinterConnections 

MapNetworkDrive RemoveNetworkDrive AddPrinterConnection 

AddWindowsPrinterConnection RemovePrinterConnection 

SetDefaultPrinter. 

Пример использования объекта WScript.Network: 

On Error Resume Next 

Set Wnet = WScript.CreateObject("WScript.Network") 

S = "Домен = " & Wnet.UserDomain & vbLF 

S = S & "Имя компьютера = " & Wnet.ComputerName & vbLF 

S = S & "Пользователь = " & Wnet.UserName 

Net_f = "\\lks\asg" 

Wnet.MapNetworkDrive "I:", Net_f,"True" 

If Err.Number <> 0 Then 'сетевая папка не найдена 
 MsgBox S & vbLF & "Cетевая папка " & Net_f & _ 

  " не найдена",vbCritical,"Ошибка..." 

End If 

Set NetDrives = Wnet.EnumNetworkDrives 

If not isObject(NetDrives) Then  

  MsgBox "Нет сетевых дисков",vbExclamation,_  

        "Нет данных..." 

Else 

 i = 0 

 S = S & vbLF & "Количество сетевых дисков = " _ 

       & NetDrives.Count/2 & vbLF 

 While i <= NetDrives.Count-1 

    S = S & NetDrives.Item(i) & " - " & _ 

NetDrives.Item(i+1) & vbLF 

    i = i+2 

 Wend 

 MsgBox S, vbInformation,_ 

  "Информация о компьютере" 

End If  

Пример работы программы 

показан на рисунке 15.1. 

 

Задание 

Напишите программу, которая определит имя Вашего компьюте-

ра, имя пользователя и покажет список его логических дисков. 

 
Рисунок 15.1. Пример работы с 

объектом WScript.Network 

Лабораторная работа № 16. Работа с объектами 
Windows OLE Automation (Microsoft ActiveX) 

Данная возможность используется в целях администрирования 

серверов, компьютеров и их сетей, имеет неограниченные возможно-

сти, и здесь подробно рассматриваться не будет. Большое количество 

примеров использования можно найти на сайте корпорации Microsoft 

(http://www.microsoft.com/technet/scriptcenter/). 

Приведем лишь примеры использования для объектов системы 

Microsoft Office. 

Пример 1. Программа открывает существующий файл системы 

Microsoft Office Word (c:\vbscripts\test_prg.doc) и добавляет в него но-

вые строки текста: 

Const END_OF_STORY = 6 

Const MOVE_SELECTION = 0 

Set objWord = CreateObject("Word.Application") 

objWord.Visible = True 

Set objDoc = _ 

  objWord.Documents.Open("c:\vbscripts\test_prg.doc") 

Set objSelection = objWord.Selection 

objSelection.EndKey END_OF_STORY, MOVE_SELECTION 

objSelection.TypeParagraph() 

objSelection.Font.Size = "14" 

objSelection.TypeText "Это первая строка, которую" & _ 

     " VBScript добавила в документ" 

objSelection.TypeParagraph() 

objSelection.TypeText "Это еще одна строка, " & _ 

" которую VBScript добавила в документ" 

objSelection.TypeParagraph() 

objSelection.TypeText "Сегодняшняя дата " & Date() 

objSelection.TypeParagraph() 

Пример работы программы показан на рисунке 16.1. 

 
Рисунок 16.1. Пример работы с объектом Word.Application 

file:///C:\ASG\�������\Informatika\���_������_�����������\����������������\VBScript\Microsoft%20VBScripr%20Reference\glossary.htm%23defOLEAutomationObject
http://www.microsoft.com/technet/scriptcenter/

Пример 2. Программа открывает новый файл системы Microsoft 

Office Excel и добавляет в его ячейки текст, числа и расчетную форму-

лу: 

Set objExcel = CreateObject("Excel.Application") 

With objExcel 

  .Visible = True 

  .Workbooks.Add 

  .Cells(1, 1).Value = "Тест работы VBScript c Excel" 

  .Cells(2, 1).Value = 1.111 

  .Cells(3, 1).Value = 2.222 

  .Cells(4, 1).Value = "=A2+A3" 

End With 

Результат работы про-

граммы показан на рисунке 

16.2. 

 

 

 

 

Пример 3. Программа устанавливает связь с базой данных систе-

мы Microsoft Office Access (используется база данных, описанная в [1], 

раздел 3.3.4), выполняет запрос на отбор данных и показывает их в ок-

не, приведенном на рисунке 16.3:  

Dim Conn, rs, S 

Set Conn = CreateObject("ADODB.Connection") 

   '1 вариант – использование провайдера программного ядра базы Access: 
Conn.Open "PROVIDER=Microsoft.Jet.OLEDB.4.0;" & _ 

          "DATA SOURCE=D:\ASG\Контингент.mdb" 

   '2 вариант – использование провайдера Access Connectivity Engine  

  ' Conn.Open "PROVIDER=Microsoft.ACE.OLEDB.12.0;" & _ 
          "DATA SOURCE=C:\AGTU\Контингент.accdb" 

   '3 вариант – использование  источника данных' пользователя, 

      ' описанного заранее в Администраторе источников 

      ' данных Windows (ODBC) 

   'Conn.Open "DSN=Контингент" '          
Set rs = CreateObject("ADODB.Recordset") 

Set rs = Conn.Execute("SELECT NZ, FIO FROM Список")  

S=" № зач.    Фамилия, имя, отчество" & vbLF 

S = S & "-------------------------------------" & vbLF 

Do Until rs.EOF 

    S = S & rs("NZ") & "  " & rs("fio") & vbLF 

    rs.MoveNext  

Loop  

 

Рисунок 16.2. Пример работы с 

объектом Excel.Application 

msgbox S,,"Данные таблицы Список"  

If rs.State = adStateOpen then 

   rs.Close 

End If 

If Conn.State = adStateOpen then 

   Conn.Close 

End If 

 

Рисунок 16.3. Пример работы с объектами 

ADODB.Connection и ADODB.Recordset 

Задание 

Напишите программу, которая выполнит следующие действия: 

a) откроет любой Ваш документ Microsoft Office Word; 

b) откроет любой Ваш документ Microsoft Office Excel; 

c) прочитает записи одной из таблиц вашей базы данных Access 

и запишет их в текстовый файл (каждая запись – отдельная 

строка). 

Лабораторная работа № 17. Использование скриптов 
на HTML-страницах 

В разделе 4.12 учебника [1] приведены некоторые сведения о 

языке гипертекстовой разметки HTML, который позволяет создавать 

сложные структурированные красиво оформленные web-страницы. 

Указывалось также, что на этих страницах могут использоваться про-

граммные компоненты – скрипты на языках сценариев, таких, как Java 

Script, VBScript и TCL для создания динамических страниц и использо-

вания HTML как средства создания сетевых приложений.  

Сценарий на странице должен располагаться внутри пары тэгов 

<SCRIPT> и </SCRIPT>. По умолчанию языком сценария считается 

JavaScript, при использовании языка VBScript следует использовать 

указание в тэге <SCRIPT LANGUAGE="VBSCRIPT">. 

Любая программа на языке приведенная выше, может исполнять-

ся и вызовом ее через Интернет-браузер, если файл программы сохра-

нять не с расширением *.vbs, а как *.htm или *.html. 

Простейший пример скрипта, который будет выполнен програм-

мой Internet Explorer, если его написать в Блокноте и сохранить с име-

нем My_script1.html:  

 <SCRIPT LANGUAGE="VBScript"> 

MsgBox ("Привет, WWW!") 

</SCRIPT> 

Открыв файл My_script1.html двойным щелч-

ком мыши (можно также открыть его, как и файлы 

*.vbs, из текстовых редакторов EmEditor и Aditor), 

мы увидим пустую web-страницу и созданную ей 

Windows-форму, показанную на рисунке 17.1. При 

этом необходимо разрешить исполнение скриптов 

на web-странице. 

Сценарий может быть расположен в любом 

месте HTML-документа, но обычно он располагается в начале или в 

конце документа. 

Пример HTML-страницы с математическими вычислениями на 

ней, выполняемыми скриптом (результат работы программы показан на 

рисунке 17.2): 

<HTML>  

  <HEAD>  

    <TITLE>Работа с VBScript</TITLE>  

   </HEAD>  

   <BODY>  

    <H3>Пример использования VBScript на HTML-

cтранице</H3>  

 
Рисунок 17.1. 

Простейший 

скрипт на   

web-странице 

    <P>Задайте количество и цену 

       и нажмите кнопку "Расчет стоимости" </P>  

    <FORM NAME="frm"> 

        <TABLE> 

           <TR> 

              <TD> Количество</TD> 

              <TD> Цена</TD> 

              <TD> Стоимость</TD> 

           </TR> 

           <TR> 

             <TD><INPUT TYPE="Text" NAME="Kolich" 

SIZE=12></TD> 

             <TD><INPUT TYPE="Text" NAME="cena" 

SIZE=12></TD> 

             <TD><INPUT TYPE="Text" NAME="stoimost" 

SIZE=12></TD> 

           </TR> 

          </TABLE> 

        <BR> 

        <INPUT TYPE ="Button" NAME="Calc" _ 

VALUE="Расчет стоимости"> 

    </FORM> 

  </BODY> 

  <SCRIPT LANGUAGE="VBScript">  

     Sub Calc_OnClick() 

         document.frm.stoimost.value  = _  

         Round(document.frm.Kolich.value _  

         * Document.frm.cena.value,2) 

      End Sub 

   </SCRIPT>  

 </HTML> 

 

 

 Задание 

Модифицируйте про-

грамму Вашей лабора-

торной работы № 4 с 

размещением VBScript-

программы на web-

странице. На форма 

должны быть поля для 

X, Y и кнопка «Рас-

чет». 

 

 

 
Рисунок 17.2. Использование VBScript на 

web-странице 

Приложение 1. Математические функции языка 

Синтаксис функции Описание 

Abs(числ.выраж.) Абсолютная величина числового выражения 

Array(сп.эл.массива) Создание массива из списка 

Atn(числ.выраж.) Арктангенс, числовое выражение в радианах 

Cbool(числ.выраж.) Возвращает True для ненулевого числового 

выражения, False для 0, ошибка для нечисло-

вого значения 

CByte(числ.выраж.) Преобразование в субтип Byte 

CCur(числ.выраж.) Преобразование в субтип Currency 

CDbl(числ.выраж.) Преобразование в субтип Double 

CInt(числ.выраж.) Преобразование в субтип Integer 

CLng(числ.выраж.) Преобразование в субтип Long 

Cos(числ.выраж.) Косинус, числовое выражение в радианах 

CSng(числ.выраж.) Преобразование в субтип Single 

CStr(числ.выраж.) Преобразование в субтип Строка 

Exp(числ.выраж.) Функция экспонента Exp(X) или e
x
 

Eval(выражение) Вычисляет выражение и возвращает его ре-

зультат 

Fix(числ.выраж.) Целая часть числа 

Int(числ.выраж.) Целая часть числа 

FormatCurrency(числ.выраж

. [,дес.зн. [,вед.ноль [,исп.() для 

отр.чисел [,исп.разд.]]]]) 

Форматирование числ.выраж. в заданный 

денежный формат с указанием денежной 

единицы, заданной в Windows 

FormatNumber(числ.выраж. 

[,дес.зн. [,вед.ноль [,исп.() для 
отр.чисел [,исп.разд.]]]]) 

Форматирование числ.выраж. в заданный 

формат 

FormatPercent(числ.выраж. 

[,дес.зн. [,вед.ноль [,исп.() для 

отр.чисел [,исп.разд.]]]]) 

Форматирование числ.выраж. в заданный 

процентный формат (/100) с символом % 

Hex(числ.выраж.) Шестнадцатеричное число десятичного чи-

слового выражения 
(числ.выраж.) \ (числ.выраж.) Деление двух числовых выражения, резуль-

татом которого является целое число (цело-

численное деление). До деления результаты 

округляются до целого числа (<0,5равно 0, 

>=0.5 равно1), у результата операции отбра-

сывается дробная часть 

Lbound(имя_массива[,размерно

сть]) 

Наименьшее возможное значение индекса в 

массиве.. Если размерность массива не ука-

зан, он одномерный 

Log(числ.выраж.) Натуральный логарифм выраже-

ния. Если нужно вычислить ло-

гарифм с основанием n, следует 

использовать выражение: 

Logn(x) = Log(x) / Log(n) 

(числ.выраж.) Mod 

(числ.выраж.) 
Остаток целочисленного деления. Перед де-

лением результаты численных выражений 

округляются до целых чисел  

Oct(число) Строка, представляющая восьмерич-

ную величину числа 
Randomize Инициализация генератора случайных чисел 

Rnd[(числ.выраж.)] Случайное число большее или равное нулю, 

но меньшее 1.  

RGB(red, green, blue) Преобразование цветового значения, где ка-

ждый из трех цветов в диапазоне от 0 до 255 

в одно число, рассчитанное по формуле: 
CLng(red + (green * 256) + ( 

blue* 65536)) 

Round(числ.выраж. [, дес.зн.]) Число, округленное до заданного количества 

десятичных знаков. Если дес.зн. не указано, 

его значение равно 0 

Sgn(числ.выраж.) Знак числа (>0 – 1, 0 – 0, <0 –  -1) 

Sin(числ.выраж.) Синус угла, выраженного в радианах 

Sqr(числ.выраж.) Квадратный корень числа 

Tan(числ.выраж.) Тангенс угла, выраженного в радианах 

TypeName(varИмя) Название типа переменной (см. далее для 
VarType) 

Ubound(имя_массива[, размер-

ность]) 
Наибольшее возможное значение индекса в 

массиве. Если размерность массива не ука-

зан, он одномерный 

VarType(varИмя) Числовое обозначение типа переменной: 
   0 vbEmpty 
   1 vbNull 
   2 vbInteger 
   3 vbLong 
   4 vbSingle 
   5 vbDouble 
   6 vbCurrency 
   7 vbDate 
   8 vbСтрока 
   9 vbObject 
  10 vbError 
  11 vbBoolean 
  12 vbVariant 
  13 vbDataObject 
  17 vbByte 
8192 vbArray 

Приложение 2. Строковые функции языка 

Синтаксис функции Описание 
Asc(строка) ANSI-код первого символа в строке 

Chr(число) Символ заданного ANSI-кода 

Eval(выражение) Вычисляет выражение и возвращает его ре-

зультат 

InStr([нач.поз., ]строка1, стро-
ка2[, тип сравн.])  

Позиция строки2 в строке1 начиная с 
нач.поз. поиска для заданного типа сравне-

ния (vbBinaryCompare или vbTextCompare, 
если не указано, то первый) 

InStrRev(строка1, строка2[, 

нач.[, тип сравн.]]) 

То же, что и  InStr, но номер позиции с кон-

ца строки 

Join(имя_массива[, раздели-
тель]) 

Строка, созданная из элементов массива  

LTrim(строка), RTrim(строка), 

Trim(строка) 

Строка без пробелов слева (LTrim), справа 

(RTrim), или без тех и других (Trim). 

LCase(строка) Преобразует все символы строки в строчные 

Left(строка, длина) Возвращает заданное количество символов с 

начала строки 

Len(строка | имя_пер.) Число символов в строке или строковой пе-
ременной 

Mid(строка, нач.[, длина]) Возвращает заданное количество символов с 

заданной позиции нач. в строке 

Replace(исх_стр, стр_поиска, 
стр_замены[, нач.[, колич.[, тип 

сравн.]]]) 

Замена в исходной строке строки поиска на 
строку замены, начиная с позиции нач., за-

данное количество раз 

Right(строка, длина) Возвращает заданное количество символов с 
конца строки 

Space(количество) Строка из заданного количества пробелов 

Split(исх_стр.[, разделитель[, 

количество[,.[, тип сравн.]]]) 

Возвращает одномерный массив строк, полу-

ченный разбиением исх_стр. по разделите-
лям на заданное количество частей. Если 

разделитель не указан, за него принимается 
знак пробела.  

StrComp(строка1, строка2[,тип 

сравн.]) 

Сравнение строк. Если строка1<строка2, 

возвращается -1, если строка1=строка2, 
возвращается 0, если строка1>строка2, воз-

вращается 1. 

String(количество, символ) Создает строку из заданного количества за-
данных символов 

StrReverse(строка) Переворачивает строку задом-наперед 

UCase(строка) Преобразует все символы строки в пропис-

ные 

Приложение 3. Функции работы с датой и временем  

Синтаксис функции Описание 

CDate(стр.выр.)  Преобразование строкового выражения в 

подтип Дата-время. 

DateAdd(формат, количество, 

<исходное дата-время>) 
 Дата-время, к которому добавлен заданное 
количество времени для заданного формата: 

"yyyy" Год 

"q" Квартал 

"m" Месяц 

"y" День года 

"d" День 

"w" день недели 

"ww" Week of year 

"h" Час 

"n" Минута 

"s" Секунда 
 

DateDiff(формат, дата1, да-

та2 [,первый_день_нед[, пер-

вая_нед_года]]) 

Интервал между двумя временными интерва-
лами в заданном формате 

DatePart(формат, дата 

[,первый_день_нед[, пер-

вая_нед_года]]) 

Часть даты в заданном формате 
 

DateSerial(год, месяц, 

день) 

 Дата для заданных года, месяца и дня 

DateValue("дата-время") Подтип Дата-время для заданного строкового 
выражения 

Day("дата-время")  День месяца – целое  число лт 1 до 31 

FormatDateTime(дата[, 

формат) 
Форматирование даты в заданный формат: 
vbGeneralDate, vbLongDate, vbShortDate, 

vbLongTime, vbShortTime 

Hour(время) Час дня, целое число от 0 до 23 

Minute(время) Минуты часа, целое число от 0 до 59 

Month(дата) День месяца,  целое число от 1 до 12 

MonthName(месяц[, 

сокр_назв]) 

Строковое обозначение месяца, сокращенное 
название (True) или полное (False). Если не 

указано, False – полное название. 

Now Текущая дата и время системных часов 

Second(время) Секунды в минуте,  целое число от 0 до 59  

Time Текущее время системных часов 

TimeSerial(час, минута, се-

кунда) 

 Время для заданных значений часа, минуты 

и секунды 

TimeValue("время")  Преобразование строкового значения вре-
мени в подтип Время  

Weekday(дата, [пер-

вый_день_нед.]) 

Целое число – день недели 

WeekdayName(день_нед., 

сокр., первый_день_нед.) 
Строка, показывающая название заданного 

дня недели  

Year(дата) Целое число – год для заданной даты 

Приложение 4. Константы даты и времени 

Константа Значение Описание 

vbSunday 1 Воскресенье 

vbMonday 2 Понедельник 

vbTuesday 3 Вторник 

vbWednesday 4 Среда 

vbThursday 5 Четверг 

vbFriday 6 Пятница 

vbSaturday 7 Суббота 

vbUseSystemDayOfWeek 0 Использовать для определения 

первого дня недели региональные 

настройки системы. 

vbFirstJan1 1 Первой неделей в году считается 

та, в которой было 1 января. 

vbFirstFourDays 2 Первой неделей в году считается 

та, в которой было по крайней 

мере четыре дня нового года. 

vbFirstFullWeek 3 Первой неделей в году считается 

первая полная неделя. 

vbGeneralDate 0 Дата и время выводятся в форма-

те, определяемом региональными 

настройками системы. 

vbLongDate 1 Выводить дату, используя пол-

ный формат. 

vbShortDate 2 Выводить дату, используя крат-

кий формат. 

vbLongTime 3 Выводить время, используя пол-

ный формат. 

vbShortTime 4 Выводить время, используя крат-

кий формат. 

Приложение 5. Логические функции и операторы 
языка 

Синтаксис функции Описание 

IsArray(переменная) True, если переменная = массив   

IsDate(выражение) True, если выражение может быть преоб-

разовано в дату 

IsEmpty(выражение) True, если выражение пустое   

IsNull(выражение) True, если выражение не содержит дан-

ных (значение Null) 

IsNumeric(выражение) True, если значение выражения – число  

IsObject(выражение) True, если выражение - объект 

Результат = 
выражение1 And выра-

жение2  

Логическая конъюнкция. Если выражения 

слева и справа от него истинны, резуль-
тат True, иначе False или Null 

Результат = выраже-

ние1 Eqv выражение2 

Проверка эквивалентности двух выражений. 

Возвращает True, если они имеют одина-
ковое значение 

Результат = выраже-
ние1 Imp выражение2 

Логическая импликация. Выражение 

E1 Imp E2 возвращает False, если 
E1 = True и E2 = False, во всех осталь-

ных случаях – True 

Результат = object1 Is 

object2  

Проверка эквивалентности двух объектов 

Результат = Not выра-

жение 

Логическое отрицание. Возвращает True, 

если условие ложно и наоборот 

Результат = 
выражение1 Or выра-

жение2 

Логическая дизъюнкция. Должно быть ис-

тинным хотя бы одно из выражений 

Результат = выраже-
ние1 Xor выражение2 

Логическое исключение. Выражение 

E1 Xor E2 возвращает True, если только 
E1 = True или только E2 = True, иначе – 

False 

file:///C:\ASG\�������\Informatika\���_������_�����������\����������������\VBScript\Microsoft%20VBScripr%20Reference\glossary.htm%23defNull

Приложение 6. Методы и свойства объекта 
ADO.Recordset 

Свойства/Коллекции 

Имя Краткое описание 

AbsolutePage  Указывает, на какой странице текущая запись находится 

AbsolutePosition  Указывает порядковую позицию текущей записи в объекте Recordset 

ActiveCommand  Указывает объект Command, созданный связанным объектом Re-
cordset  

ActiveConnection  Указывает, какому объекту  Connection в настоящее время принад-
лежат указанная команда, Recordset, или запись 

BOF, EOF  BOF — указывает, что текущая позиция - перед первой записью в 
объекте Recordset.  

EOF — указывает, что текущая позиция - после последней записи 

Bookmark  Закладка, которая однозначно определяет текущую запись в объекте 
Recordset или устанавливает текущую запись в объекте Recordset на 

запись, идентифицированную этой значением закладки Bookmark 

CacheSize  Указывает число записей объекта Recordset, которые кэшируются в 
памяти 

CursorLocation  Указывает местоположение курсора 

CursorType  Указывает тип курсора, используемого в объекте Recordset 

DataMember  Указывает Имя компонента данных, который будет найден для объ-
екта, на который ссылается свойство DataSource  

DataSource  Указывает объект, который содержит данные, которые будут пред-
ставлены как объект Recordset 

EditMode  Указывает статус редактирования текущей записи 

Fields (Collection) Коллекция полей для записи Recordset 

Filter  Указывает фильтр данных для Recordset. 

Index  Указывает имя индекса, заданного для объекта Recordset 

LockType  Указывает тип блокировки 

MarshalOptions  Указывает, какие записи должны быть возвращены назад на сервер 

MaxRecords  Указывает максимальное количество записей, которое возвращает 
Recordset для запроса 

PageCount  Указывает сколько страниц данных содержит объект Record-
set 

PageSize  Указывает из сколько записей состоит одна логическая стра-

ница данных в Recordset. 

Свойства 
(Collection) 

Коллекция свойств объекта 

RecordCount  Указывает количество записей в объекте Recordset 

Sort  Указывает одно имя поля или несколько по которым объект 
Recordset сортируется в порядке возрастания или убывания 

Source  Указывает источник данных для объекта Recordset 

State  Состояние объекта: открыт, закрыт; соединяется, исполняет-
ся, выполняется 

Status  Статус текущей записи 

StayInSync  Указывает, в иерархическом объекте Recordset, выполняются или 
нет изменения соответствующих дочерних записей, если позиция 
родительской строки изменяется  

ms-help://MS.MSACCESS.DEV.12.1049/MSACCESS.DEV/content/HV01228447.htm
ms-help://MS.MSACCESS.DEV.12.1049/MSACCESS.DEV/content/HV01228449.htm
ms-help://MS.MSACCESS.DEV.12.1049/MSACCESS.DEV/content/HV01228463.htm
ms-help://MS.MSACCESS.DEV.12.1049/MSACCESS.DEV/content/HV01228463.htm
ms-help://MS.MSACCESS.DEV.12.1049/MSACCESS.DEV/content/HV01228463.htm
ms-help://MS.MSACCESS.DEV.12.1049/MSACCESS.DEV/content/HV01228463.htm
ms-help://MS.MSACCESS.DEV.12.1049/MSACCESS.DEV/content/HV01228463.htm

Методы 

Имя Краткое описание 

AddNew  Создание новой записи в обновляемом объекте Recordset. 

Cancel  Отменить выполнение  

CancelBatch  Отменить выполнение пакетного обновления 

CancelUpdate  Отменить обновление для объекта Recordset или коллекции 

полей или объекта Record до вызова метода Update 

Clone  Создать дубликат объекта Recordset, доступный только 

для чтения 

Close  Закрыть объект и все подчиненные объекты 

CompareBookmarks  Сравнивает две закладки и возвращает их относительное 

положение (CompareEnum) 

Delete  Удаляет текущую запись или группу записей  

Find  Поиск записи в объекте Recordset для заданных условий 

GetRows  Переносит множество записей объекта Recordset в массив 

GetString  Возвращает Recordset как строку с разделителями 

Move  Перемещает текущую позицию в объекте Recordset 

MoveFirst, Move-
Last, MoveNext, 
and MovePrevious 

Перемещает текущую позицию в объекте Recordset в на-

чало, в конец, на следующую или предыдущую запись 

NextRecordset  Очищает текущий объект Recordset и возвращает следующий Re-
cordset при продвижении через ряд команд 

Open  Открыть курсор  

Requery  Обновляет данные в объекте Recordset перезапуском запроса, на 
котором базируется этот объект 

Resync  Обновляет данные в текущем объекте Recordset, или коллекции 
полей объекта из основной базы данных  

Save  Сохраняет Recordset в файле, или объекте Stream 

Seek  Ищет индекс Recordset, чтобы быстро определить запись, которая 
соответствует указанным значениям, и изменяет текущую пози-
цию строки в объекте Recordsetѐ 

Supports  Определяет, поддерживает ли указанный объект Recordset спе-
цифический тип функциональности 

Update  Сохраняет любые изменения, которые Вы делаете в текущей 

строке объекта Recordset, или коллекции полей объекта Record 

UpdateBatch  Записывает все пакетные обновления на диск 

События (events) 

Имя Краткое описание 

EndOfRecordset  Происходит, когда выполняется попытка двигаться за последнюю 
запись объекта Recordset 

FetchComplete  Происходит после того, как все записи в длинной асинхронной 
операции были найдены в Recordset 

FetchProgress  Происходит периодически в течение длинной асинхронной опера-
ции, чтобы сообщить, сколько записей было найдено в операции 
выборки 

WillChangeField 
and FieldChange-
Complete  

Событие WillChangeField происходит прежде, чем изменяетcя зна-
чение одного или более полей в Recordset. Событие FieldChange-
Complete происходит после того, как значение одного или более 
полей изменилось 

WillChangeRecord Событие WillChangeRecord происходит перед изменением одной 

ms-help://MS.MSACCESS.DEV.12.1049/MSACCESS.DEV/content/HV01228463.htm
ms-help://MS.MSACCESS.DEV.12.1049/MSACCESS.DEV/content/HV01228470.htm
ms-help://MS.MSACCESS.DEV.12.1049/MSACCESS.DEV/content/HV01228463.htm
ms-help://MS.MSACCESS.DEV.12.1049/MSACCESS.DEV/content/HV01228463.htm
ms-help://MS.MSACCESS.DEV.12.1049/MSACCESS.DEV/content/HV01228463.htm
ms-help://MS.MSACCESS.DEV.12.1049/MSACCESS.DEV/content/HV01228470.htm

and RecordChange-
Complete  

или более записей в Recordset. Событие RecordChangeComplete 
происходит после одного или более изменений записей 

WillChangeRecord-
set and Recordset-
ChangeComplete  

Событие WillChangeRecordset происходит прежде, чем изменяется 
Recordset. Событие RecordsetChangeComplete происходит после 
того, как Recordset изменился 

WillMove and 
MoveComplete  

Событие WillMove происходит прежде, чем изменяется текущая 
позиция в Recordset. Событие MoveComplete происходит после 
изменения текущей позиции в Recordset 

WillMove and 
MoveComplete  

Событие WillMove происходит прежде, чем изменяется текущая 
позиция в Recordset. Событие MoveComplete происходит после 
изменения текущей позиции в Recordset 

